


Computer
Shop

Alfred Görgens

Schneider CPC 464/664
Anwenderprogramme 1



Birkhäuser

B

Computer Shop
Band 32

Alfred Görgens

Schneider
CPC 464/664
Anwenderprogramme 1

Birkhäuser Verlag
Basel · Boston · Stuttgart

CIP-Kurztitelaufnahme der Deutschen Bibliothek

Görgens, Alfred:

Schneider CPC 464, 664 : Anwenderprogramme / Alfred Görgens. –
Basel ; Boston ; Stuttgart : Birkhäuser

1 (1985).

(Computer-Shop ; Bd. 32)

ISBN 3-7643-1725-6

NE: GT

Die vorliegende Publikation ist urheberrechtlich geschützt.
Alle Rechte vorbehalten. Kein Teil dieses Buches darf ohne schriftliche
Genehmigung des Verlages in irgendeiner Form durch Fotokopie, Mikrofilm,
Kassetten oder andere Verfahren reproduziert werden. Auch die Rechte der
Wiedergabe durch Vortrag, Funk und Fernsehen bleiben vorbehalten.

© 1985 Birkhäuser Verlag, Basel

Umschlaggestaltung: Bruckmann & Partner, Basel

Printed in Germany

ISBN 3-7643-1725-6

Inhalt

Vorwort	7
Archiv	9
Bauen & Wohnen	16
Terminkalender	28
KlarText	33
Drucker-Zeichen	46
Zeichensatz-Generator	55
Balkendiagramm	63
Angebots-Plotter	71
Key-Maker	79
Adressen-Verwaltung	81
Anwender-Entspannung	92
Peripheres	115

Vorwort

Was sind Anwenderprogramme? - Da müssen wir uns erst einmal ganz dumm stellen und sagen: Anwenderprogramme sind computer-gesteuerte Hilfsroutinen, die durch menüorientierte Ablauf-funktionen über mehrheitlich miteinander korrespondierende Subroutinen benutzerspezifische Arbeitsaufgaben bewältigen, deren Ergebnisse über Bildschirm oder Peripheriegeräte ausgegeben werden. Etwas komplizierter ausgedrückt: Anwenderprogramme helfen die Arbeit zu erleichtern.

In diesem Buch finden Sie eine ganze Reihe von komfortablen Anwenderprogrammen mit verschiedensten Einsatzmöglichkeiten. Mit einem universellen Archiv-Programm lassen sich z.B. im häuslichen Bereich Videofilme, Briefmarken, Bücher oder Urlaubsfotos ordnen oder im beruflichen Bereich Zeitschriften-artikel (für Journalisten), Baupläne (für Architekten) u.v.m. Durch bestimmte Suchroutinen sind einzelne Datensätze abruf-bar, so daß mühsames Durchforsten von Karteikarten erspart bleibt.

Besonders interessant für alle, die viel schreiben müssen, ist das Textverarbeitungsprogramm KlarText, mit dem Sie kürzere Texte (Aufsätze, Briefe, Rundschreiben usw.) erstellen und be-arbeiten können. Trotz der reinen BASIC-Fassung sind Funktio-nen wie Textblock verschieben, Textblock kopieren, wortweises vor- und zurückfahren ohne weiteres möglich.

Für alle, die statistische Werte gerne in farbige Grafik um-setzen wollen, ist das Anwenderprogramm Balkendiagramm geeig-net. Ebenso ist das Programm Angebots-Plotter professionell einsetzbar. Mit ihm lassen sich Skizzen aller Art (z.B. für Bilderrahmen, Metallwaren, elektronische Bauteile u.ä.) anfer-tigen und nach Eingabe von Stückzahl und Einzelpreis die Kaufsumme automatisch ausrechnen.

Besonders komfortabel ist das Anwenderprogramm Adressen-Ver-waltung. Neben einfachem Erfassen von Anschriften lassen sich Dateien nach bestimmten Begriffen durchsuchen oder Adressen mit Brieftext mischen.

Darüber hinaus gibt es in einem Kapitel "Anwender-Entspannung"

einige unterhaltsame Spielchen für zwischendurch. Hier ist besonders das Programm Musik-Box hervorzuheben, mit dem auf bequeme Weise Noten eingegeben, abgespeichert und gespielt werden können.


Als Abschluß finden Sie noch einen Anhang mit nützlichen Hinweisen für Peripheriegeräte, damit Sie immer guten Anschluß für Ihren Schneider finden.


Archiv

Alles muß seine Ordnung haben. So dachten die Leute schon früher und füllten ganze Häuser mit Regalen, zwischen denen fleißige, kleine Männlein mit Ärmelschonern und Nickelbrillen hin- und herflitzten, um irgendwelche Bücher, Stadtpläne, Landkarten oder Geburtsurkunden einzuordnen. Und heute? Ja heute ist das alles ganz anders. Da schiebt man eine Diskette in die Floppy, da tippt man RUN und da passiert - gar nichts.

Sie haben nämlich vergessen, das folgende Archivprogramm einzugeben. Sie können damit praktisch alles ordnen, was Sie sammeln; z.B. Videofilme, Zeitschriftenartikel, Micky-Maus-Hefte, Weinsorten, Spiele oder Briefmarken. Das vorliegende Programm ist als Bücherarchiv eingerichtet. Aber es versteht sich von selbst, daß Sie anstelle von "Autor", "Titel", "Verlag" usw. auch "Zeitschrift", "Heft-Nummer", "Anbaugbiet" o.ä. eingeben können. Die Programmstruktur bleibt dabei unverändert.

Die Bedienung des Archivprogramms ist einfach; alle wichtigen Anweisungen werden auf dem Bildschirm angezeigt.





```

10 'Schneider Archivprogramm
20 GOSUB 1400
30 MODE 1
40 WINDOW#2,1,40,24,25
50 LOCATE 2,2:PRINT"*****
*****"
60 LOCATE 2,3:PRINT"*****  H A U P M E N Ü
*****"
70 LOCATE 2,4:PRINT"*****
*****"
80 LOCATE 1,7:PRINT"1. Neue Datei erstellen"
90 LOCATE 1,9:PRINT"2. Datei einlesen"
100 LOCATE 1,11:PRINT"3. Datei ausdrucken"
110 LOCATE 1,13:PRINT"4. Dateinsätze selektieren"
120 LOCATE 1,15:PRINT"5. Einzelne Datensätze ausdruc
ken"
130 LOCATE 1,17:PRINT"6. Programm beenden"
140 PRINT#2,CHR$(12)
150 IF DN$="" THEN PRINT#2,"Keine Datei in Bearbeitu
ng" ELSE PRINT#2,"Aktuelle Datei: ";DN$
160 LOCATE 4,20:INPUT "Ihre Eingabe bitte";M
170 ON M GOTO 190,390,600,800,1100,180
180 CLS:END

```

```

190 'Neue Datei erstellen
200 PRINT#2,CHR$(12)
210 INPUT#2,"Eingabe Dateiname: ";DN$
220 MODE 2
230 WINDOW#2,1,80,24,25
240 PRINT#2,"Aktuelle Datei: ";DN$
250 PRINT#2,"Datensätze werden 2K-Blockweise übertragen"
260 OPENOUT DN$+".DAT"
270 INPUT"Autor           :";AU$
280 INPUT"Titel          :";T$
290 INPUT"Verlag         :";V$
300 INPUT"Ersch.Jahr/Preis :";EP$
310 INPUT"Bemerkungen    :";BM$
320 PRINT#9,AU$:PRINT#9,T$:PRINT#9,V$:PRINT#9,EP$:PRINT#9,BM$
330 PRINT:PRINT
340 PRINT"Weiterarbeiten mit <W>  Zum Hauptmenü mit <M>"
350 W$=INKEY$:IF W$="" THEN GOTO 350
360 IF W$="W" OR W$="w" THEN PRINT:PRINT:GOTO 270
370 IF W$="M" OR W$="m" THEN CLOSEOUT:GOTO 30
380 GOTO 350
390 'Datei einlesen
400 PRINT#2,CHR$(12)
410 INPUT#2,"Eingabe Dateiname: ";DN$
420 MODE 2
430 WINDOW#2,1,80,24,25
440 PRINT#2,"Aktuelle Datei: ";DN$
450 PRINT#2,"Datei wird eingelesen"
460 OPENIN DN$+".DAT"
470 IF EOF THEN GOTO 570
480 INPUT#9,AU$,T$,V$,EP$,BM$
490 PRINT"Autor           :";AU$
500 PRINT"Titel          :";T$
510 PRINT"Verlag         :";V$
520 PRINT"Ersch.Jahr/Preis :";EP$
530 PRINT"Bemerkungen    :";BM$
540 PRINT:PRINT
550 INPUT "Weiter mit RETURN";A$
560 GOTO 470
570 CLOSEIN
580 LOCATE 1,25:PRINT"Ende der Datei. Zum Menü mit RETURN";
590 INPUT A$:GOTO 30
600 'Datei ausdrucken
610 PRINT#2,CHR$(12)

```

```

620 INPUT#2,"Eingabe Dateiname: ";DN$
630 MODE 2
640 WINDOW#2,1,80,24,25
650 PRINT#2,"Aktuelle Datei: ";DN$
660 PRINT#2,"Datei wird eingelesen und gedruckt"
670 OPENIN DN$+".DAT"
680 IF EOF THEN 770
690 INPUT#9,AU$,T$,V$,EP$,BM$
700 LPRINT"Autor           :";AU$
710 LPRINT"Titel           :";T$
720 LPRINT"Verlag          :";V$
730 LPRINT"Ersch.Jahr/Preis: ";EP$
740 LPRINT"Bemerkungen      :";BM$
750 LPRINT:LPRINT
760 GOTO 680
770 CLOSEIN
780 LOCATE 1,25:PRINT"Ende der Datei. Zum Menü mit R
ETURN";
790 INPUT A$:GOTO 30
800 'Datensätze selektieren
810 PRINT#2,CHR$(12)
820 INPUT#2,"Eingabe Dateiname: ";DN$
830 MODE 2
840 WINDOW#2,1,80,24,25
850 PRINT#2,"Aktuelle Datei: ";DN$
860 INPUT#2,"Welches Wort soll selektiert werden";SW
$
870 PRINT#2,CHR$(12)
880 PRINT#2,"Datei wird eingelesen und geprüft"
890 PRINT#2,"Einen Moment Geduld bitte";
900 SW=LEN(SW$)
910 OPENIN DN$+".DAT"
920 IF EOF THEN 1070
930 INPUT#9,AU$,T$,V$,EP$,BM$
940 IF LEFT$(AU$,SW)=SW$ THEN 1000
950 IF LEFT$(T$,SW)=SW$ THEN 1000
960 IF LEFT$(V$,SW)=SW$ THEN 1000
970 IF LEFT$(EP$,SW)=SW$ THEN 1000
980 IF LEFT$(BM$,SW)=SW$ THEN 1000
990 GOTO 920
1000 PRINT"Autor           :";AU$
1010 PRINT"Titel           :";T$
1020 PRINT"Verlag          :";V$
1030 PRINT"Ersch.Jahr/Preis: ";EP$
1040 PRINT"Bemerkungen      :";BM$
1050 PRINT:LPRINT
1060 GOTO 920

```

```

1070 CLOSEIN
1080 LOCATE 1,25:PRINT"Ende der Datei. Zum Menü mit
RETURN";
1090 INPUT A$:GOTO 30
1100 'Einzelne Datensätze ausdrucken
1110 PRINT#2,CHR$(12)
1120 INPUT#2,"Eingabe Dateiname: ";DN$
1130 MODE 2
1140 WINDOW#2,1,80,24,25
1150 PRINT#2,"Aktuelle Datei: ";DN$
1160 INPUT#2,"Welches Wort soll selektiert werden";S
W$
1170 PRINT#2,CHR$(12)
1180 PRINT#2,"Datei wird eingelesen und geprüft"
1190 PRINT#2,"Einen Moment Geduld bitte";
1200 SW=LEN(SW$)
1210 OPENIN DN$+".DAT"
1220 IF EOF THEN 1370
1230 INPUT#9,AU$,T$,V$,EP$,BM$
1240 IF LEFT$(AU$,SW)=SW$ THEN 1300
1250 IF LEFT$(T$,SW)=SW$ THEN 1300
1260 IF LEFT$(V$,SW)=SW$ THEN 1300
1270 IF LEFT$(EP$,SW)=SW$ THEN 1300
1280 IF LEFT$(BM$,SW)=SW$ THEN 1300
1290 GOTO 1220
1300 LPRINT"Autor           :";AU$
1310 LPRINT"Titel           :";T$
1320 LPRINT"Verlag          :";V$
1330 LPRINT"Ersch. Jahr/Preis: ";EP$
1340 LPRINT"Bemerkungen        :";BM$
1350 LPRINT:LPRINT
1360 GOTO 1220
1370 CLOSEIN
1380 LOCATE 1,25:PRINT"Ende der Datei. Zum Menü mit
RETURN";
1390 INPUT A$:GOTO 30
1400 SYMBOL AFTER 91
1410 SYMBOL 91,102,24,60,102,126,102,102,0
1420 SYMBOL 92,195,60,102,102,102,102,60,0
1430 SYMBOL 93,102,0,102,102,102,102,60,0
1440 SYMBOL 123,108,0,120,12,124,204,118,0
1450 SYMBOL 124,0,102,0,60,102,102,60,0
1460 SYMBOL 125,0,102,0,102,102,102,62,0
1470 SYMBOL 126,60,102,102,124,102,102,108,224
1480 RETURN

```

30: Mit MODE 1 wird der Textmodus gewählt, der 40 mal 25 Zeichen bietet (MODUS 2 bestimmt den Textmodus mit 80 mal 25 Zeichen).

40: Beim Schneider können Sie bis zu acht verschiedene Windows (Fenster) definieren. Diese lassen sich bequem für Programm-erklärungen, Menüs oder andere Texteinblendungen nutzen. In diesem Fall wird ein Window (mit #2) in den untersten beiden Bildschirmzeilen definiert, wo durch PRINT#2, "... " alle Bedie-nungsfunktionen unabhängig vom übrigen Bildschirminhalt einge-blendet werden.

50 bis 130: Mit dem LOCATE-Befehl können Sie den Cursor an beliebige Stellen auf dem Bildschirm positionieren. In diesen Zeilen wird das Hauptmenü auf den Bildschirm geschrieben.

140: Das definierte Bildschirmfenster wird gelöscht.

150: Solange noch kein Dateiname definiert ist, erscheint im Window #2 die Meldung "Keine Datei in Bearbeitung".

160 und 170: ME wurde als Denkhilfe für "Menüeingabe" genom-men. Je nachdem, welche Zahl Sie drücken, springt der Rechner zu der entsprechenden Zeilennummer; also bei 1=190, bei 2=390 usw.

210 bis 230: Eingabe des gewünschten Dateinamens und Umschal-ten in MODE 2. Dadurch muß auch das Window neu definiert werden.

260: Das Programm wird in dieser Version mit dem CPC-Dis-kettenlaufwerk betrieben. Bei Kassettenbetrieb kann der Zusatz .DAT fortfallen.

320 bis 380: Nach der Eingabe eines Datenblocks wird jedesmal gefragt, ob Sie weiterarbeiten oder zum Menü zurückkehren möchten.

390 bis 460: Angaben zum Einlesen einer Datei.

470: Wenn eine Datei bereits zu Ende ist, der Rechner aber noch weitere Datensätze lesen soll, tritt ein Fehler auf. Durch die Anweisung IF EOF THEN 570 kann man vorher bestim-men, in welcher Zeile das Programm in diesem Fall fortgesetzt werden soll, ohne daß eine Unterbrechung auftritt.

600 bis 790: Wie 390 bis 590, jedoch Ausgabe mit dem Drucker.

800: Eine Besonderheit bei diesem relativ kurzen Anwenderprogramm ist die Möglichkeit, abgespeicherte Dateien nach bestimmten Begriffen zu durchsuchen. Diese Funktion läßt sich sogar durchführen, wenn Sie gar nicht wissen, welchen Begriff Sie suchen: Stellen Sie sich vor, Sie hätten in diesem Bücherarchiv mehrere Titel eines Autors gespeichert, der "irgendwie mit M beginnt". Geben Sie einfach M als Suchwort ein, dann werden alle Eingaben, die mit M beginnen, auf dem Bildschirm angezeigt.


860 bis 900: Nachdem Sie das Suchwort eingegeben haben (Zeile 860), wird mit LEN(SW\$) seine Länge ermittelt.

930 bis 980: Die einzelnen Datensätze werden in Zeile 930 eingelesen und hier mit dem Suchwort verglichen. Sobald ein Datenstring mit dem gesuchten Begriff übereinstimmt, erfolgt die Ausgabe auf dem Bildschirm. Im anderen Fall wird der nächste Datensatz eingelesen.

1070 bis 1090: Sobald eine Datei "durchgeschnüffelt" wurde, können Sie zum Hauptmenü zurückkehren.

1100 bis 1390: Einzelne Datensätze selektieren wie in den Zeilen 800 bis 1090, jedoch mit Druckerausgabe.


1400 bis 1480: Durch die SYMBOL-Anweisung können in Verbindung mit SYMBOL AFTER die Zeichen bestimmter Tasten umdefiniert werden. In diesem Fall wurden die Tasten mit den ASCII-Codes 91 bis 93 und 123 bis 126 zu deutschen Umlauten umdefiniert (entspr. dem internationalen Code zur Druckeransteuerung). Sie können also ganz normale Eingaben mit Umlauten machen, die ohne weitere Umwandlungen auch auf jedem gängigen Drucker erscheinen. Die Tastatur ist während des Programms folgendermaßen belegt:




Bauen & Wohnen

Nach A wie "Archiv" folgt B wie "Bauen & Wohnen". Falls Sie also ein Eigentum bauen, kaufen oder modernisieren wollen, können Sie mit dem nachfolgenden Anwenderprogramm die günstigste Finanzierung (über Bausparvertrag, Hypothek oder normalen Kredit) ausrechnen lassen. Außerdem läßt sich ermitteln, welche Rendite ein erworbenes Objekt bringt. Wahlweise werden alle Ergebnisse auf dem Bildschirm oder per Drucker ausgegeben.


Das Programm ist modular aufgebaut; d.h. wenn Sie z.B. nur die Jahreszinsen, die Jahrestilgung und die verbleibende Restschuld bei Aufnahme einer Hypothek zur Finanzierung eines Objektes interessieren, können Sie sich beim Abtippen des Listings auf dieses eine Modul beschränken. Das Programm wird fehlerfrei funktionieren.





Vom Hauptmenü aus wählen Sie die verschiedenen Unterprogramme, indem Sie die jeweilige Zahl eingeben.



Mit dem Programm-Modul "Bausparen" wird der Tilgungs- und Zinsverlauf eines Bauspardarlehens ermittelt. Geben Sie dazu die Höhe des Darlehens, die monatliche Rate, den effektiven Jahreszins und die Laufzeit des Darlehens ein; natürlich nur die reinen Zahlen, nicht "DM" oder "%".




Im Unterprogramm "Normal sparen" wird der Sparverlauf aufgezeigt, der bei einer bestimmten Sparsumme und n Einzahlungen pro Jahr in einem bestimmten Zeitraum und Jahreszins entsteht. Geben Sie die monatliche Sparsumme, die Anzahl der Einlagen pro Jahr, die Anzahl der Sparjahre und den Zinssatz ein.



Das Programm-Modul "Hypothek" ermittelt die Jahreszinsen, die Jahrestilgung und die verbleibende Restschuld einer Hypothek. Außerdem wird die gesamte Laufzeit und die gesamte Zinssumme angezeigt. Geben Sie hier die Hypothek-Summe, den Zinssatz, die jährliche Tilgung und die Einzahlungen pro Jahr ein.

Mit dem Unterprogramm "Abschreibung" können Sie den Wertverlust einer Anschaffungssumme bzw. die jährliche Abschreibung für ein Objekt ermitteln lassen. Geben Sie die Kaufsumme des Objektes und die Abschreibungsrate ein. Der Rest wird vom Rechner besorgt.



Das letzte Programm-Modul heißt "Rendite" und errechnet die Jahresrendite (in Prozent) eines Objektes, das zu einem bestimmten Preis erworben wurde. Geben Sie die Kaufsumme, den effektiven Nennwert des Objektes, den erwarteten Jahresertrag und die gewünschte Laufzeit ein.

```

10 'Bauen & Wohnen
20 GOSUB 2140
30 MODE 1
40 WINDOW#2,1,40,24,25:C$=CHR$(137)
50 U$="#####.##"
60 LOCATE 2,1:PRINT"*****
*****"
70 LOCATE 2,2:PRINT"*****  H A U P T M E Ü  ***
*****"
80 LOCATE 2,3:PRINT"*****
*****"
90 LOCATE 5,6:PRINT"1. Bausparen"
100 LOCATE 5,8:PRINT"2. Normal Sparen"
110 LOCATE 5,10:PRINT"3. Hypothek"
120 LOCATE 5,12:PRINT"4. Abschreibung"
130 LOCATE 5,14:PRINT"5. Rendite berechnen"
140 LOCATE 5,16:PRINT"6. Programm beenden"
150 INPUT#2,"Ihre Eingabe bitte ";E
160 ON E GOTO 180,620,910,1350,1600,170
170 CLS:END
180 'Bausparen (Bildschirmausgabe)
190 MODE 2
200 WINDOW#1,1,80,1,3:WINDOW#2,1,80,5,20:WINDOW#3,1,
80,21,25
210 PRINT#1,"*****
*****"
220 PRINT#1,"*****  P r o g r a m m - M o d
u l   B a u s p a r e n  *****"
230 PRINT#1,"*****
*****";
240 PRINT#2,CHR$(12):PRINT#3,CHR$(12)
250 INPUT#2,"Eingabe Bauspardarlehens-Summe: ";BD
260 INPUT#2,"Eingabe der Monatsrate      : ";MR
270 INPUT#2,"Eingabe Jahreszins in %     : ";JZ
280 INPUT#2,"Eingabe der Laufzeit in Jahren: ";LZ
290 PRINT#2
300 INPUT#2,"Ausgabe mit Drucker (J/N)";P$
310 PRINT#2,CHR$(12)
320 PRINT#2,"Darlehen: ";BD;"  Monatsrate: ";MR;"  Zin
s: ";JZ;"  Laufzeit: ";LZ
330 LOCATE 6,10:PRINT"Jahr"
340 LOCATE 17,10:PRINT"Zinsbetrag"
350 LOCATE 32,10:PRINT"Tilgung"
360 LOCATE 47,10:PRINT"Restbetrag"
370 IF P$="J" OR P$="j" THEN GOSUB 1810
380 JZ=JZ/100/12
390 J=0:B=0

```

```

400 TILG=BD:C=0
410 FOR X=1 TO LZ
420 C=C+BD*JZ:BD=BD-MR+BD*JZ
430 IF BD>0 THEN NEXT X:GOTO 450
440 BD=0
450 J=J+1
460 A1=INT(100*C+0.5)/100:A2=INT(100*(TILG-BD)+0.5)/
100:A3=INT(100*BD+0.5)/100
470 LOCATE 6,11+J:PRINT J;
480 LOCATE 17,11+J:PRINT USING U$;A1;
490 LOCATE 32,11+J:PRINT USING U$;A2;
500 LOCATE 47,11+J:PRINT USING U$;A3
510 IF P$="J" OR P$="j" THEN GOSUB 1870
520 B=B+C
530 IF BD>0 THEN 400
540 PRINT#2:PRINT#2:PRINT#2
550 PRINT#3,CHR$(12)
560 PRINT#3,"Gesamte Zinssumme: ";USING U$;B;
570 PRINT#3,"      Laufzeit: ";J-1;"Jahre und";X;"Mona
te"
580 IF P$="J" OR P$="j" THEN GOSUB 1890
590 PRINT#3:INPUT#3,"Weiterrechnen = R  Zum Menü = M
";P$
600 IF P$="R" OR P$="r" THEN 240
610 GOTO 30
620 'Normal sparen (Bildschirmausgabe)
630 MODE 2
640 WINDOW#1,1,80,1,3:WINDOW#2,1,80,5,20:WINDOW#3,1,
80,21,25
650 PRINT#1,"*****
*****"
660 PRINT#1,"*****  Programm - Modul
Normal sparen *****"
670 PRINT#1,"*****
*****";
680 PRINT#2,CHR$(12):PRINT#3,CHR$(12)
690 INPUT#2,"Eingabe Sparsumme      ":";SP
700 INPUT#2,"Anzahl der Einlagen   ":";AE
710 INPUT#2,"Anzahl der Sparjahre  ":";SJ
720 INPUT#2,"Eingabe Zinssatz      ":";JZ
730 PRINT#2
740 INPUT#2,"Ausgabe mit Drucker (J/N)";P$
750 PRINT#2,CHR$(12)
760 PRINT#2,"Sparsumme: ";SP;" Jahreseinlagen: ";AE;"
Sparjahre: ";SJ;" Zinssatz: ";JZ
770 LOCATE 10,8:PRINT"Jahr":LOCATE 20,8:PRINT"Kapita
lbildung"

```

```

780 IF P$="J" OR P$="j" THEN GOSUB 1920
790 JZ=1+JZ/AE/100
800 A=0
810 FOR Y=1 TO SJ:FOR X=1 TO AE
820 A=(A+SP)*JZ:NEXT X
830 A1=INT(100*A+0.5)/100
840 LOCATE 10,9+Y:PRINT Y
850 LOCATE 20,9+Y:PRINT USING U$;A1
860 IF P$="J" OR P$="j" THEN GOSUB 1960
870 NEXT Y:PRINT#2:PRINT#2:PRINT#2
880 PRINT#3:INPUT#3,"Weiterrechnen = R   Zum Menü = M
";P$
890 IF P$="R" OR P$="r" THEN 680
900 GOTO 30
910 'Hypothek (Bildschirmausgabe)
920 MODE 2
930 WINDOW#1,1,80,1,3:WINDOW#2,1,80,5,20:WINDOW#3,1,
80,21,25
940 PRINT#1,"*****
*****"
950 PRINT#1,"*****   Programm - Mod
u l   Hypothek   *****"
960 PRINT#1,"*****
*****";
970 PRINT#2,CHR$(12):PRINT#3,CHR$(12)
980 INPUT#2,"Eingabe Hypotheksumme           :";HS
990 INPUT#2,"Eingabe Zinssatz in %           :";JZ
1000 INPUT#2,"Tilgung                         :";TILG
1010 INPUT#2,"Einzahlungen pro Jahr          :";ZA
1020 PRINT#2
1030 INPUT#2,"Ausgabe mit Drucker (J/N)";P$
1040 PRINT#2,CHR$(12)
1050 PRINT#2,"Summe: ";HS;"   Zinssatz: ";JZ;"   Tilgung
: ";TILG;"   Zahlungen: ";ZA
1060 IF P$="J" OR P$="j" THEN GOSUB 1970
1070 LOCATE 5,8:PRINT"Jahr"
1080 LOCATE 20,8:PRINT"Zinsen"
1090 LOCATE 35,8:PRINT"Tilgung"
1100 LOCATE 50,8:PRINT"Restkapital"
1110 JZ=JZ/100/ZA
1120 J=0:A=0
1130 B=INT(100*HS*(JZ+TILG/100/ZA)+0.5)/100
1140 TS=HS:X=0
1150 FOR Y=1 TO ZA
1160 X=X+HS*JZ:HS=HS+HS*JZ-B
1170 IF HS>0 THEN NEXT Y:GOTO 1190
1180 HS=0

```

```

1190 J=J+1
1200 A1=INT(100*X+0.5)/100
1210 A2=INT(100*(TS-HS)+0.5)/100
1220 A3=INT(100*HS+0.5)/100
1230 LOCATE 5,9+J:PRINT J;
1240 LOCATE 20,9+J:PRINT USING U$;A1;
1250 LOCATE 35,9+J:PRINT USING U$;A2;
1260 LOCATE 50,9+J:PRINT USING U$;A3
1270 IF P$="J" OR P$="j" THEN GOSUB 2020
1280 A=A+X
1290 IF HS>0 THEN 1140
1300 PRINT#3,"Zu zahlen DM";B;" pro Jahr; ";ZA;" mal
"
1310 IF P$="J" OR P$="j" THEN GOSUB 2030
1320 PRINT#3:INPUT#3,"Weiterrechnen = R zum Menü =
M";P$
1330 IF P$="R" OR P$="r" THEN 970
1340 GOTO 30
1350 'Abschreibung (Bildschirmausgabe)
1360 MODE 2
1370 WINDOW#1,1,80,1,3:WINDOW#2,1,80,5,20:WINDOW#3,1
,80,21,25
1380 PRINT#1,"*****
*****"
1390 PRINT#1,"***** Programm - Modu
l Abschreibung *****"
1400 PRINT#1,"*****
*****";
1410 PRINT#2,CHR$(12):PRINT#3,CHR$(12)
1420 INPUT#2,"Eingabe Kaufsumme      ";KO
1430 INPUT#2,"Abschreibungsrate     ";AS
1440 PRINT#2:INPUT#2,"Ausgabe mit Drucker (J/N)";P$
1450 PRINT#2,CHR$(12):PRINT#2,"Kaufsumme: ";KO;" Abs
chreibungsrate: ";AS
1460 LOCATE 10,8:PRINT"Restwert"
1470 LOCATE 30,8:PRINT"Abschreibung"
1480 IF P$="J" OR P$="j" THEN GOSUB 2040
1490 A=KO:J=0
1500 A=A-KO*AS/100
1510 IF A>=0 THEN 1530
1520 A=0
1530 LOCATE 10,9+J:PRINT A
1540 LOCATE 30,9+J:PRINT KO-A
1550 IF P$="J" OR P$="j" THEN GOSUB 2080
1560 IF A>0 THEN J=J+1:GOTO 1500
1570 INPUT#3,"Weiterrechnen = R Zum Menü = M";P$
1580 IF P$="R" OR P$="r" THEN 1410

```

```

1590 GOTO 30
1600 'Rendite (Bildschirmausgabe)
1610 MODE 2
1620 WINDOW#1,1,80,1,3:WINDOW#2,1,80,5,20:WINDOW#3,1
,80,21,25
1630 PRINT#1,"*****
*****"
1640 PRINT#1,"*****  Programm - M o
d u l   R e n d i t e   *****"
1650 PRINT#1,"*****
*****";
1660 PRINT#2,CHR$(12):PRINT#3,CHR$(12)
1670 INPUT#2,"Eingabe Kaufsumme      :";KO
1680 INPUT#2,"Nennwert des Objektes  :";NW
1690 INPUT#2,"Jahresertrag in %      :";JZ
1700 INPUT#2,"Laufzeit in Jahren     :";LZ
1710 PRINT#2:INPUT#2,"Ausgabe mit Drucker (J/N)";P$
1720 PRINT#2,CHR$(12)
1730 PRINT#2,"Kaufsumme: ";KO; " Nennwert: ";NW; " Jahre
sertrag: ";JZ; "% Laufzeit: ";LZ
1740 IF P$="J" OR P$="j" THEN GOSUB 2090
1750 A1=NW*JZ/KO+100*(NW-KO)/(KO*LZ)
1760 PRINT#3,"Jahresrendite in %: ";USING U$;A1
1770 IF P$="J" OR P$="j" THEN GOSUB 2120
1780 PRINT#3:INPUT#3,"Weiterrechnen = R Zum Menü =
M";P$
1790 IF P$="R" OR P$="r" THEN 1660
1800 GOTO 30
1810 'Bausparen (Druckerausgabe)
1820 PRINT#8
1830 PRINT#8,"Darlehen: ";BD; " Monatsrate: ";MR; " Zi
ns: ";JZ; " Laufzeit: ";LZ
1840 PRINT#8
1850 PRINT#8,C$; "Jahr";C$; "Zinsbetrag";C$; "Tilgung";
C$; "Restbetrag"
1860 PRINT#8:RETURN
1870 PRINT#8,C$; J;C$; A1;C$; A2;C$; A3
1880 RETURN
1890 PRINT#8:PRINT#8,"Gesamte Zinssumme: ";USING U$;B
1900 PRINT#8,"Laufzeit: ";J-1; "Jahre und"; X; "Monate"
1910 RETURN
1920 'Normal sparen (Druckerausgabe)
1930 PRINT#8,"Sparsumme: ";SP; " Jahreseinlagen: ";AE;
" Sparjahre: ";SJ; " Zinssatz: ";JZ
1940 PRINT#8:PRINT#8,c$; "Jahr";C$; "Kapitalbildung"
1950 PRINT#8:RETURN
1960 PRINT#8,C$; Y;C$; A1:RETURN

```

```

1970 'Hypothek (Druckerausgabe)
1980 PRINT#8,"Summe:";HS;" Zinssatz:";JZ;" Tilgung
:";TILG;" Zahlungen:";ZA
1990 PRINT#8
2000 PRINT#8,"Jahr";C$;"Zinsen";C$;C$;"Tilgung";C$;C
$;"Restkapital"
2010 PRINT#8:RETURN
2020 PRINT#8,J;C$;A1;C$;A2;C$;A3:RETURN
2030 PRINT#8,"Zu zahlen DM";B;" pro Jahr; ";ZA;" mal
":RETURN
2040 PRINT#8,CHR$(12):PRINT#2,"Kaufsumme:";KO;" Abs
chreibungsrate:";AS
2050 PRINT#8:PRINT#8
2060 PRINT#8,"Restwert";C$;"Abschreibung"
2070 PRINT#8:RETURN
2080 PRINT#8,A;C$;KO-A:RETURN
2090 'Rendite (Druckerausgabe)
2100 PRINT#8,"Kaufsumme:";KO;" Nennwert:";NW;" Jahre
sertrag:";JZ;"% Laufzeit:";LZ
2110 PRINT#8:PRINT#8:RETURN
2120 PRINT#8,"Jahresrendite in %:";USING U$;A1
2130 PRINT#8:RETURN
2140 SYMBOL AFTER 91
2150 SYMBOL 91,102,24,60,102,126,102,102,0
2160 SYMBOL 92,195,60,102,102,102,102,60,0
2170 SYMBOL 93,102,0,102,102,102,102,60,0
2180 SYMBOL 123,108,0,120,12,124,204,118,0
2190 SYMBOL 124,0,102,0,60,102,102,60,0
2200 SYMBOL 125,0,102,0,102,102,102,62,0
2210 SYMBOL 126,60,102,102,124,102,102,108,224
2220 RETURN

```

50: Beim späteren Ausdruck der Zahlen entstehen z.T. Tabellen. Damit die Dezimalstellen jeweils untereinander stehen, wird der PRINT USING-Befehl eingesetzt. Das CPC-BASIC erlaubt eine ganze Reihe von komfortablen Möglichkeiten, PRINT USING anzuwenden. Einziger Nachteil: Das USING-Format muß für jeden Ausdruck einzeln angegeben werden; d.h. Sie müßten bei jeder PRINT-Anweisung den Zusatz USING #####.##;x programmieren. Um dies zu erleichtern, wird hier das Format in eine Variable gesteckt, so daß Sie nur jeweils PRINT USING U\$;x programmieren müssen. - Die Variable C\$=CHR\$(137) dient als Tabulatur-Sprunganweisung für den Drucker bei der Ausgabe der Tabellen. 137 ist der Code für EPSON FX-80. Schauen Sie bitte in Ihrer Bedienungsanleitung nach, welchen Code Ihr Drucker hierfür benutzt.

160: Je nach Eingabe springt der Computer in die angegebene Zeile; bei 1=180, bei 2=620 usw. Falls Sie einzelne Module fortlassen, müssen Sie die Sprungbefehle hier entsprechend ändern.

170: Wenn Sie im Hauptmenü eine 6 eingeben, wird das Programm beendet.

210 bis 290: Bildschirmausgabe für das erste Programm-Modul.

300: Wenn Sie nach der Frage "Ausgabe mit Drucker" ein J oder j gedrückt haben, springt der Rechner vor der weiteren Ausführung des Programms zur Zeile 1810, wo eine Druckerausgabe erfolgt, die mit der Bildschirmausgabe identisch ist. Wenn Sie keine Druckerausgabe wünschen, drücken Sie einfach die RETURN-Taste.

380 bis 530: Hier werden die einzelnen Eingaben verarbeitet. Vom Darlehen (BD) wird die Bausparrate (MR) abgezogen und der Zinsanteil (JZ) addiert. Der Vorgang wird so lange wiederholt, bis die Restsumme = 0 ist.

590 bis 610: Nachdem die gesamte Zinssumme und die Laufzeit errechnet ist, können Sie zum Hauptmenü zurückkehren oder in diesem Programmteil weitere Eingaben machen.

620 bis 780: Eingabemenü für das zweite Modul.

790 bis 870: Zur Ermittlung der Sparkapital-Summe wird gemäß der eingegebenen Anzahl der Einlagen pro Jahr die Sparsumme zur Sparrate addiert und mit dem Zinssatz multipliziert. Die Anzahl der Sparjahre bestimmt die FOR...NEXT-Durchläufe in Zeile 810.

940 bis 1100: Eingabemenü für das dritte Modul.

1110 bis 1310: Hier werden die Fälligkeiten, die jährlich aufzubringen sind und der Zins- und Tilgungsanteil für die Hypothek ausgerechnet und ausgegeben.

1350 bis 1480: Eingabemenü für das vierte Modul.

1490 bis 1560: Ermittlung der Abschreibungssummen und des Restwertes eines Objektes.

1600 bis 1740: Eingabemenü für das fünfte Modul.

1750: Zur Ermittlung der Rendite für ein Objekt wird zunächst der Nennwert (NW) mit dem erwarteten Jahresertrag (JZ) multipliziert und durch die Kaufsumme (KO) dividiert. Dieses Ergebnis wird addiert zur Summe aus Nennwert minus Kaufpreis, geteilt durch Kaufpreis mal Laufzeit (LZ).


1810 bis 2130: Unterprogramme für die Druckerausgaben.

2140 bis 2220: Definition der deutschen Umlaute, wie auf Seite 15 dargestellt.

Terminkalender

Mit dem folgenden Anwenderprogramm können Sie sich verschiedene Terminkalender für ein Jahr oder einen bestimmten Monat aufstellen und ausdrucken lassen (z.B. einen für alle Geburtstage, einen für wichtige Messen usw.). Alle geplanten Monate können einzeln abgespeichert und beliebig aufgerufen werden. Zusätzlich lassen sich alle Bildschirmausgaben auch ausdrucken.

Die vorliegende Programmversion bezieht sich auf das Jahr 1985. Sie können sich den Kalender jedoch auf alle folgenden (oder zurückliegenden) Jahre umschreiben; dazu müssen Sie nur die jeweils letzte Zahl in den DATA-Zeilen ändern. Diese Zahl kennzeichnet den Wochentag, mit dem der Monat beginnt. Im Jahr 1985 begann z.B. der Januar mit dem Dienstag, also dem zweiten Wochentag (s. Zeile 390). 1986 beginnt der Januar an einem Mittwoch, also dem dritten Wochentag. Sie müßten in diesem Fall die Zahl 2 durch 3 ersetzen.



***** September 1985 *****

Sonntag	1	
Montag	2	
Dienstag	3	Franz Kremer Geb.
Mittwoch	4	
Donnerstag	5	
Freitag	6	
Samstag	7	Birkhäuser Buch kaufen
Sonntag	8	
Montag	9	
Dienstag	10	
Mittwoch	11	
Donnerstag	12	
Freitag	13	Hochzeitstag
Samstag	14	
Sonntag	15	
Montag	16	Annemarie Geb.
Dienstag	17	
Mittwoch	18	
Donnerstag	19	
Freitag	20	Computer kaufen
Samstag	21	
Sonntag	22	
Montag	23	
Dienstag	24	
Mittwoch	25	
Donnerstag	26	
Freitag	27	Stephan Geb.
Samstag	28	
Sonntag	29	
Montag	30	

Ausdruckbeispiel für den Monat September

```
10 'Terminkalender
```

```
20 GOSUB 510:MODE 2:WINDOW#1,1,80,1,3:WINDOW#2,1,80,5  
,23:WINDOW#3,1,80,24,25:PRINT#2,CHR$(12):PRINT#3,CHR$(  
12)
```

```
30 DIM T$(7),TM$(31):T$(1)="Montag":T$(2)="Dienstag":  
T$(3)="Mittwoch":T$(4)="Donnerstag":T$(5)="Freitag":T
```

```

$(6)="Samstag":T$(7)="Sonntag"
40 PRINT#1,"*****"
*****"
50 PRINT#1,"***** T E R M I N K A
L E N D E R *****"
60 PRINT#1,"*****"
*****"
70 PRINT#2,CHR$(12):PRINT#3,CHR$(12)
80 INPUT#2,"Eingabe des gesuchten Monats (nur die ers
ten drei Buchstaben): ";M$
90 PRINT#2
100 INPUT#2,"Eingabe des Termin datums (nur Zahl)
: ";D
110 PRINT#2:PRINT#2:INPUT#2,"Ausgabe mit Drucker (J/N
)";P$
120 INPUT#2,"Laden von Diskette (J/N)";LA$
130 INPUT#2,"Speichern auf Diskette (J/N)";SP$
140 IF SP$="J" OR SP$="j" THEN OPENOUT M$
150 RESTORE:X=1
160 READ A$,B,C:B$=LEFT$(A$,3)
170 IF LEFT$(M$,3)=B$ THEN 200
180 X=X+1:IF X>12 THEN PRINT#3,"Eingabefehler";CHR$(7
):FOR Z=1 TO 1000:NEXT Z:GOTO 70
190 GOTO 160
200 PRINT#2,CHR$(12):X=1:Y=1
210 A$="***** "+A$+" 1985 *****"
220 IF P$="J" OR P$="j" THEN PRINT#8,A$:PRINT#8
230 IF LA$="J" OR LA$="j" THEN OPENIN M$:INPUT#9,A$:P
RINT#2,A$:GOTO 600
240 IF SP$="J" OR SP$="j" THEN PRINT#9,A$
250 PRINT#2,A$:PRINT#2
260 IF X=D THEN INPUT#3,"Eingabe des Termins (Text) "
;TM$(Y):INPUT#3,"Weitere Termine gewünscht (J/N)";X$
270 IF X$="J" OR X$="j" THEN PRINT#3,CHR$(12):INPUT#3
,"Eingabe Termin datum (nur Zahl): ";D:X$=""
280 IF X<16 THEN PRINT#2,T$(C),X;TM$(Y)
290 IF X>15 THEN LOCATE 40,X-9:PRINT T$(C),X;TM$(Y)
300 IF P$="J" OR P$="j" THEN PRINT#8,T$(C),X;TM$(Y)
310 IF SP$="J" OR SP$="j" THEN PRINT#9,T$(C):PRINT#9,
X:PRINT#9,TM$(Y)
320 C=C+1:IF C=8 THEN C=1
330 X=X+1:IF X>8 THEN 350
340 Y=Y+1:GOTO 260
350 PRINT#3,CHR$(12)
360 INPUT#3,"Weitere Monate planen (J/N)";X$
370 IF X$="J" OR X$="j" THEN RUN

```

```

380 MODE 1:END
390 DATA Januar,31,2
400 DATA Februar,28,5
410 DATA März,31,5
420 DATA April,30,1
430 DATA Mai,31,3
440 DATA Juni,30,6
450 DATA Juli,31,1
460 DATA August,31,4
470 DATA September,30,7
480 DATA Oktober,31,2
490 DATA November,30,5
500 DATA Dezember,31,7
510 SYMBOL AFTER 91
520 SYMBOL 91,102,24,60,102,126,102,102,0
530 SYMBOL 92,195,60,102,102,102,102,60,0
540 SYMBOL 93,102,0,102,102,102,102,60,0
550 SYMBOL 123,108,0,120,12,124,204,118,0
560 SYMBOL 124,0,102,0,60,102,102,60,0
570 SYMBOL 125,0,102,0,102,102,102,62,0
580 SYMBOL 126,60,102,102,124,102,102,108,224
590 RETURN
600 INPUT#9,T$(C),X,TM$(Y):IF X<16 THEN PRINT#2,T$(C)
,X;TM$(Y)
610 IF X>15 THEN LOCATE 40,X-9:PRINT T$(C),X;TM$(Y)
620 IF P$="J" OR P$="j" THEN PRINT#8,T$(C),X;TM$(Y)
630 C=C+1:IF C=8 THEN C=1
640 X=X+1:Y=Y+1:IF EOF THEN 650 ELSE GOTO 600
650 CLOSEIN:GOTO 350

```

30: Jedem Index der DIMensionierten Variable T\$ wird ein Wochentag zugewiesen.

40 bis 140: Das Eingabemenü auf dem Bildschirm. Da im weiteren Programmverlauf nur die ersten drei Buchstaben des eingegebenen Monats mit den DATAs verglichen werden, müssen Sie auch lediglich drei Buchstaben eingeben. Achten Sie dabei auf korrekte Schreibweise, also z.B. "Jan" und nicht "JAN". Als nächstes geben Sie das Datum ein, an dem der erste Termin eingetragen werden soll. Der Rechner druckt zunächst alle Tage bis zum eingegebenen Datum aus. Dann wird das Programm für Ihre Termineingabe gestoppt.

150 bis 170: Die Daten ab Zeile 390 werden eingelesen. Anschließend wird verglichen, ob Ihre Eingabe mit B\$ übereinstimmt. Wenn dies der Fall ist, springt der Rechner zur Zeile 200.

180: X dient als Hilfsvariable. Sobald 12 Monate eingelesen wurden und der gesuchte Monat nicht darunter war, liegt ein Eingabefehler vor. Das Programm beginnt dann wieder in Zeile 70.

200 und 210: Sobald der eingegebene Monat gefunden ist, wird das Programm hier fortgesetzt. Als erstes erfolgt der Ausdruck des Monats und Jahres. X und Y sind Hilfsvariablen.

260: Sobald die Zählvariable X den Wert des eingegebenen Termindatums erreicht hat, wird das Programm unterbrochen, damit Sie den Termin eintragen können. Danach folgt die Frage, ob Sie weitere Eintragungen machen möchten. Geben Sie hier ein "J" für Ja oder drücken Sie für "Nein" einfach die RETURN-Taste.

270 bis 290: Wenn Sie bei der Frage nach weiteren Terminen "J" oder "j" gedrückt haben, müssen Sie hier das Datum eingeben. Der Rechner druckt dann den Kalender bis zu diesem Tag weiter aus.

320: Hier werden die Wochentage gezählt. Sobald die Zahl 7 überschritten ist (Sonntag), nimmt C wieder 1 an (Montag).

330: B ist die Variable, mit der die Länge eines Monats festgelegt wird; z.B. 31 Tage für Januar, 28 Tage für Februar. Sobald die Zählvariable X den Wert von B überschritten hat, wird das Programm in Zeile 350 fortgesetzt.

360 und 370: Nach dem Ausdruck eines Monats folgt die Abfrage, ob noch weitere Termine geplant werden sollen. Bei Eingabe eines "J" oder "j" beginnt ein neuer Programmdurchlauf.

390 bis 500: Hier stehen die Daten, die mit Zeile 160 eingelesen werden: Monat, Anzahl der Tage und Wochentag, mit dem der Monat beginnt. Wenn Sie also den Kalender auf andere Jahre umschreiben wollen, müssen Sie lediglich die Angaben für die Wochentage ändern (sowie die Monatstage für Februar, falls es sich um ein Schaltjahr handelt).


510 bis 590: Definition der deutschen Umlaute, so wie auf Seite 15 gezeigt.

600 bis 650: Unterprogramm zum Ausdruck von gespeicherten Termindaten.

KlarText

Textverarbeitung per Computer ist und bleibt das attraktivste Thema für private und halbprofessionelle Anwender. Das folgende KlarText-Programm ist ein reines BASIC-Textsystem und kann demnach keine hohen Ansprüche befriedigen. Für Briefe, Rechnungen, Schularbeiten oder andere Kurztexte ist es jedoch ausreichend. Der Speicher nimmt bis zu 12672 Zeichen auf (ca. 4 DIN A-4 Seiten). Es ist jedoch auch möglich, mehrere Textdateien lückenlos hintereinander zu drucken.

KlarText bietet eine Reihe von Steuerfunktionen, die sich beim praktischen Arbeiten als besonders nützlich erwiesen haben; z.B. eine Top-Funktion, mit der Sie zum Textanfang gelangen und eine Bottom-Funktion, die Sie zum Textende bringt. Darüber hinaus ist das "Blättern" auf dem Bildschirm möglich.



Im Hauptmenü können Sie neun verschiedene Unterprogramme auswählen. Das Programm ist für den Betrieb mit Diskettenlauf-

Probedatei mit KlarText

Dieses Textverarbeitungsprogramm ist auf den Drucker EPSON FX-80 abgestimmt. Wenn jedoch die Steuersequenzen für Schriftart, usw. geändert werden, läßt sich KlarText mit jedem anderen Drucker betreiben.

Es ist zu beachten, daß Sätze, in denen Kommas vorkommen, mit Anführungsstrichen beginnen und enden sollten, da es sonst zu der Fehlermeldung <Redo from start> kommen kann.

Man kann Fett drucken
oder auch gedehnt
Komprierte Schrift ist ebenfalls möglich.
Außerdem lassen sich Texte unterstreichen.

Bedienung von KlarText

- m oder M Zum Hauptmenü
- e oder E Editor (Änderungen)
- t oder T Top (zum Textanfang)
- b oder B Bottom (zum Textende)
- r oder R 15 Zeilen aufwärts
- c oder C 15 Zeilen abwärts
- x oder X Weiterarbeiten mit neuer Satznummer

- k oder K Bearbeiten von Textblöcken:
 - k = Kopieren
 - l = Löschen
 - v = Verschieben

Deutsche Umlaute wie Abb. Seite 15

Beginnen Sie eine Texteingabe mit Anführungsstrichen, wenn Sie Kommas verwenden und ohne, wenn Sie innerhalb des Textes Anführungsstriche verwenden.

Beenden Sie eine Textdatei grundsätzlich mit .ende oder .ENDE (wird nicht mit ausgedruckt).

Punktcommandos zur Druckersteuerung

(nur Kleinbuchstaben verwenden
und eine Leerstelle hinter dem
Kommando lassen)

- .pb Fettdruck (Bold Printing)
 - .pd Doppeldruck (Expanded Printing)
 - .ps Schmaldruck (Condensed Printing)
 - .pu Text unterstreichen
 - .lh Zeilenspationierung mit n/72 Zoll
(stets zwei Zahlen folgen lassen
z.B. .lh 04 oder .lh 12)
 - .pl Seitenlänge (Page Length) in Zeilen
(stets zwei Zahlen folgen lassen
z.B. .pl 50)
 - .pv Vertikaler Vorschub um n Zeilen
(z.B. .pv 07 oder .pv 11)
- .ende oder .ENDE Dateiendeerkennung
(muß unbedingt eingegeben werden)

Anwendungsbeispiel

```
0 .pb Dieser Text wird fett gedruckt.
1 .pv 04
2 "Da in diesem Satz ein Komma vorkommt,"
3 muß er mit Anführungsstrichen eingegeben werden.
4 .lh 12
5 .pu Dieser Satz wird in neuer Zeilen-
6 .pu spationierung mit Unterstreichung gedruckt.
7 Alle Punktcommandos gelten nur für eine Zeile.
8
9 Wird anstelle eines Textes z.B. m oder M
10 "eingegeben, gelangt man zum Hauptmenü."
11 ?m
```

```

10 'KlarText
20 GOSUB 2130
30 MODE 2
40 X=0: DIM T$(99): TY=8: SP=6: RR=80: LR=8
50 WINDOW#0,1,80,1,23: WINDOW#1,1,80,24,25
60 GOTO 1980
70 'Texteingabe
80 PRINT#0,X;: INPUT#0,T$(X)
90 IF T$(X)="M" OR T$(X)="m" THEN PRINT#1,CHR$(12): G
OTO 1980
100 IF T$(X)="E" OR T$(X)="e" THEN 180
110 IF T$(X)="T" OR T$(X)="t" THEN 270
120 IF T$(X)="B" OR T$(X)="b" THEN 320
130 IF T$(X)="K" OR T$(X)="k" THEN 380
140 IF T$(X)="R" OR T$(X)="r" THEN 660
150 IF T$(X)="C" OR T$(X)="c" THEN 770
160 IF T$(X)="X" OR T$(X)="x" THEN PRINT#1,CHR$(12):
GOTO 560
170 X=X+1: GOTO 80
180 'Editor
190 PRINT#1,CHR$(12)
200 INPUT#1,"Welcher Satz soll editiert werden";E
210 PRINT#1,CHR$(12)
220 PRINT#1,T$(E)
230 PRINT#1,E;: INPUT#1,T$(E)
240 INPUT#1,"Weiter editieren (J oder RETURN = Nein)
";E$
250 IF E$="" THEN PRINT#1,CHR$(12): GOTO 960
260 GOTO 180
270 'Top
280 PRINT#0,CHR$(12)
290 FOR Z=0 TO 15: PRINT#0,Z;T$(Z): NEXT Z
300 PRINT#0
310 GOTO 70
320 'Bottom
330 PRINT#0,CHR$(12)
340 IF X>14 THEN X1=X-15
350 FOR Z=X1 TO X: PRINT#0,Z;T$(Z): NEXT Z
360 PRINT#0
370 GOTO 70
380 'Blockbearbeitung
390 PRINT#1,CHR$(12)
400 INPUT#1,"Blockbearbeitung: K=Kopieren V=Verschi
eben L=Löschen. Eingabe";E$
410 IF E$="K" OR E$="k" THEN 450
420 IF E$="V" OR E$="v" THEN 520
430 IF E$="L" OR E$="l" THEN 590

```

```

440 PRINT#1,CHR$(12):GOTO 960
450 INPUT#1,"Kopieren von Satz-Nr.";B1:INPUT#1," b
is Satz-Nr.";B2:INPUT#1,"Kopierbeginn bei Satz-Nr.";
B3
460 PRINT#1,CHR$(12)
470 PRINT#1,"Kopiervorgang. Bitte warten"
480 FOR Z=B1 TO B2:T$(B3)=T$(Z):PRINT#0,B3;T$(B3):B3
=B3+1:NEXT Z
490 INPUT#1,"Mit welcher Satznummer weiterarbeiten";
X
500 PRINT#1,CHR$(12)
510 GOTO 960
520 INPUT#1,"Verschieben von Satz-Nr.";B1:INPUT#1,"
bis Satz-Nr.";B2:INPUT#1,"Neubeginn bei Satz-Nr.";
B3
530 PRINT#1,CHR$(12)
540 PRINT#1,"Blockverschiebung. Bitte warten"
550 FOR Z=B1 TO B2:T$(B3)=T$(Z):PRINT#0,B3;T$(B3):T$(
Z)="" :B3=B3+1:NEXT Z
560 INPUT#1,"Mit welcher Satznummer weiterarbeiten";
X
570 PRINT#1,CHR$(12)
580 GOTO 960
590 INPUT#1,"Löschen von Satz-Nr.";B1:INPUT#1," bi
s Satz-Nr.";B2
600 PRINT#1,CHR$(12)
610 PRINT#1,"Löschvorgang. Bitte warten"
620 FOR Z=B1 TO B2:T$(Z)="" :NEXT Z
630 INPUT#1,"Mit welcher Satznummer weiterarbeiten";
X
640 PRINT#1,CHR$(12)
650 GOTO 960
660 'Rollen aufwärts
670 PRINT#0,CHR$(12)
680 IF X>29 THEN X1=X-15 ELSE X1=0
690 FOR Z=X1 TO X1+15:PRINT#0,Z;T$(Z):NEXT Z
700 PRINT#1,CHR$(12)
710 INPUT#1,"Weiter aufwärts (RETURN = Ja)";E$
720 IF E$<>"" THEN PRINT#1,CHR$(12):GOTO 960
730 X1=X1-15:PRINT#0,CHR$(12)
740 ON ERROR GOTO 760
750 GOTO 690
760 PRINT#1,CHR$(12):GOTO 960
770 'Rollen abwärts
780 PRINT#0,CHR$(12)
790 IF X>29 THEN X1=X+15 ELSE X1=0
800 FOR Z=X1 TO X1+15:PRINT#0,Z;T$(Z):NEXT Z

```

```

810 PRINT#1,CHR$(12)
820 INPUT#1,"Weiter abwärts (RETURN = Ja)";E$
830 IF E$<>" " THEN PRINT#1,CHR$(12):GOTO 960
840 X1=X1+15:PRINT#0,CHR$(12)
850 ON ERROR GOTO 870
860 GOTO 800
870 PRINT#1,CHR$(12):GOTO 960
880 'Zum aktuellen Text
890 PRINT#0,CHR$(12):PRINT#1,CHR$(12)
900 FOR Z=0 TO X:PRINT#0,Z;T$(Z):NEXT Z
910 GOTO 960
920 'Neue Datei eröffnen
950 INPUT#1,"Eingabe Dateiname: ";DN$:PRINT#0,CHR$(12)
):PRINT#1,CHR$(12)
960 PRINT#1,"  R----!----!----!----!----!----!----!
----!----!----!----!----!----!----!----!
970 PRINT#1,"  M=Menü E=Edit T=Top B=Bott K=Block
ck R=Up C=Down X=Neu Datei: ";DN$;
980 GOTO 70
990 'Datei abspeichern
1000 PRINT#1,CHR$(12)
1010 PRINT#1,"Eingabe Dateiname oder RETURN, wenn Da
teiname = ";DN$;
1020 INPUT#1,E$
1030 IF E$<>" " THEN DN$=E$
1040 PRINT#1,CHR$(12)
1050 PRINT#1,"Datei ";DN$;" wird gespeichert. Bitte
Warten.";
1060 OPENOUT DN$+".DAT"
1070 FOR Z=0 TO X:PRINT#9,T$(Z):NEXT Z
1080 CLOSEOUT
1090 PRINT#1,CHR$(12)
1100 PRINT#1,"Datei ";DN$;" ist abgespeichert."
1110 GOTO 2090
1120 'Datei einladen
1130 INPUT#1,"Eingabe Dateiname: ";DN$
1140 PRINT#1,CHR$(12)
1150 PRINT#1,"Datei ";DN$;" wird geladen. Bitte Wart
en.";
1160 X=0
1170 OPENIN DN$+".DAT"
1180 INPUT#9,T$(X)
1190 IF T$(X)=" .ENDE" OR T$(X)=" .ende" THEN CLOSEIN:
X=X+1:GOTO 1210
1200 X=X+1:GOTO 1180
1210 PRINT#1,CHR$(12)
1220 PRINT#1,"Datei ";DN$;" ist eingeladen."

```

```

1230 GOTO 2090
1240 `Datei von Diskette löschen
1260 PRINT#1,CHR$(12):INPUT#1,"Eingabe Dateiname, de
r gelöscht werden soll: ";E$
1270 PRINT#1,"Trägt der Dateiname den Zusatz .DAT od
er .BAK?"
1280 INPUT#1,"Geben Sie .DAT oder .BAK ein oder drüc
ken Sie RETURN ";X$
1290 E$=E$+X$
1300 öERA,5E$: ö=chr$(124), jedoch undefiniert
1310 PRINT#1,CHR$(12)
1320 PRINT#1,"Datei ";E$;" wurde gelöscht."
1330 GOTO 2090
1340 `Speicher löschen
1345 PRINT#1,CHR$(12)
1350 INPUT#1,"Gesamten Speicher löschen? Sind Sie si
cher (J oder RETURN, wenn Nein)";E$
1360 IF E$="" THEN PRINT#1,CHR$(12):GOTO 2090
1370 PRINT#1,CHR$(12)
1380 PRINT#1,"Speicher wird gelöscht. Bitte warten."
1390 FOR Z=0 TO X:T$(Z)="" :NEXT Z
1400 X=0
1410 PRINT#1,CHR$(12)
1420 PRINT#1,"Speicher wurde gelöscht."
1430 GOTO 2090
1440 `Datei ausdrucken
1450 PRINT#1,CHR$(12)
1460 INPUT#1,"Standard-Druckerformatierung (RETURN =
Ja)";E$
1470 IF E$="" THEN GOSUB 1900 ELSE GOTO 1680
1480 INPUT#1,"Stop nach jeder Seite (J oder RETURN =
Nein)";E$
1490 IF E$="J" OR E$="j" THEN ff=55 ELSE ff=1000
1500 PRINT#1,CHR$(12)
1510 PRINT#1,"Datei wird gedruckt. Bitte warten."
1520 Z=0
1530 IF LEFT$(T$(Z),3)=".pb" THEN PRINT#8,CHR$(27);"
E";RIGHT$(T$(Z),LEN(T$(Z))-4);CHR$(27);"F":GOTO 1640
1540 IF LEFT$(T$(Z),3)=".pd" THEN PRINT#8,CHR$(14);R
IGHT$(T$(Z),LEN(T$(Z))-4):GOTO 1640
1550 IF LEFT$(T$(Z),3)=".ps" THEN PRINT#8,CHR$(15);R
IGHT$(T$(Z),LEN(T$(Z))-4);CHR$(18):GOTO 1640
1560 IF LEFT$(T$(Z),3)=".pu" THEN PRINT#8,CHR$(27);"
-";CHR$(1);RIGHT$(T$(Z),LEN(T$(Z))-4);CHR$(27);"-" ;C
HR$(0):GOTO 1640
1570 IF T$(Z)=".pa" THEN PRINT#8,CHR$(12):GOTO 1640
1580 IF LEFT$(T$(Z),3)=".lh" THEN PRINT#8,CHR$(27);"

```

```

A";CHR$(VAL(RIGHT$(T$(Z),2)):GOTO 1640
1590 IF LEFT$(T$(Z),3)=".p1" THEN PRINT#8,CHR$(27);"
C";CHR$(VAL(RIGHT$(T$(Z),2)):GOTO 1640
1600 Z2=VAL(RIGHT$(T$(Z),2)):IF LEFT$(T$(Z),3)=".pv"
THEN FOR Z1=1 TO Z2:PRINT#8:NEXT Z1:GOTO 1640
1610 IF Z=ff OR Z=(2*ff) OR Z=(3*ff) THEN PRINT#1,CH
R$(12):PRINT#8,CHR$(12):INPUT#1,"Neues Blatt einlege
n. Weiter mit RETURN";E$
1620 IF T$(Z)=".ende" OR T$(Z)=".ENDE" THEN 1650
1630 PRINT#8,T$(Z)
1640 Z=Z+1:IF Z=X THEN 1650 ELSE 1530
1650 PRINT#1,CHR$(12)
1660 PRINT#1,"Datei ";DN$;" ist ausgedruckt."
1670 GOTO 2090
1680 'Drucker neu formatieren
1690 PRINT#0,CHR$(12):PRINT#1,CHR$(12)
1700 PRINT#0,"*****
*****"
1710 PRINT#0,"***** D R U C K E R A
N S T E U E R U N G *****"
1720 PRINT#0,"*****
*****"
1730 PRINT#0:PRINT#0
1740 PRINT#0,"1. Derzeitige Schrifttype: ES
C;!;chr$(";TY;)"
1750 PRINT#0,"2. Derzeitige Zeilenspationierung ES
C;A;chr$(";SP;)"
1760 PRINT#0,"3. Rechter Rand beginnt in Spalte ES
C;Q;chr$(";RR;)"
1770 PRINT#0,"4. Linker Rand endet in Spalte ES
C;l;chr$(";LR;)"
1780 PRINT#0,"5. Rucksetzen aller Sequenzen auf Urspr
ungswerte"
1790 PRINT#0
1800 PRINT#0,"6. Drucker formatieren ohne Ausdruck e
iner Datei"
1810 PRINT#0,"7. Zuruck zum Hauptmenu"
1820 INPUT#1,"Ihre Eingabe bitte:";E$
1830 IF E$="1" THEN INPUT#1,"Eingabe der neuen Schri
fttype";TY:GOTO 1680
1840 IF E$="2" THEN INPUT#1,"Eingabe der neuen Zeile
nspationierung";SP:GOTO 1680
1850 IF E$="3" THEN INPUT#1,"Eingabe der rechten Ran
dspalte";RR:GOTO 1680
1860 IF E$="4" THEN INPUT#1,"Eingabe der linken Rand
spalte";LR:GOTO 1680
1870 IF E$="5" THEN TY=8:SP=6:RR=80:LR=8:GOTO 1680

```

```

1880 IF E$="6" THEN GOSUB 1900:GOTO 1680
1890 IF E$="7" THEN PRINT#1,CHR$(12):GOTO 1980
1900 'Standard-Druckerformatierung
1910 X$=CHR$(27)+"!"+CHR$(TY)+CHR$(27)+"A"+CHR$(SP)
1920 X$=X$+CHR$(27)+"Q"+CHR$(RR)+CHR$(27)+"1"+CHR$(L
R)
1930 PRINT#8:PRINT#8,X$:RETURN
1940 'Disketteninhalt
1950 PRINT#1,CHR$(12)
1960 CAT
1970 GOTO 2090
1980 'Hauptmenü
1990 CLS
2000 PRINT#0,"*****
*****"
2010 PRINT#0,"***** K L A R T E X T
H A U P T M E N Ü *****"
2020 PRINT#0,"*****
*****"
2030 PRINT#0:PRINT#0
2040 PRINT#0,"1. Zum aktuellen Text 6.
Speicher löschen"
2050 PRINT#0,"2. Neue Datei eröffnen 7.
Datei ausdrucken"
2060 PRINT#0,"3. Datei abspeichern 8.
Drucker neu formatieren"
2070 PRINT#0,"4. Datei einladen 9.
Disketteninhalt"
2080 PRINT#0,"5. Datei von Diskette löschen E
Programm abbrechen"
2090 INPUT#1,"Ihre Eingabe bitte: ";E$
2100 IF E$="E" OR E$="e" THEN MODE 2:END
2110 E=VAL(E$)
2120 ON E GOTO 880,920,990,1120,1240,1340,1440,1680,
1940
2130 SYMBOL AFTER 91
2140 SYMBOL 91,102,24,60,102,126,102,102,0
2150 SYMBOL 92,195,60,102,102,102,102,60,0
2160 SYMBOL 93,102,0,102,102,102,102,60,0
2170 SYMBOL 123,108,0,120,12,124,204,118,0
2180 SYMBOL 124,0,102,0,60,102,102,60,0
2190 SYMBOL 125,0,102,0,102,102,102,62,0
2200 SYMBOL 126,60,102,102,124,102,102,108,224
2210 RETURN

```

40: X ist die Zählvariable für die Textstrings, die mit T\$(X) eingegeben werden. TY=8, SP=6 usw. bestimmen die Standardwerte zur Druckeransteuerung. Wenn Sie nicht den EPSON FX-80 benutzen, müssen Sie hier ggf. andere Angaben machen. Da der Schneider-Computer grundsätzlich ein Doppel-Linefeed (doppelter Zeilenvorschub) zum Drucker sendet, wird mit der Ansteuerung ESC;"A";chr\$(SP) die normale Zeilenspationierung (Dezimalwert 12) halbiert (Dezimalwert 6). Sollte Ihr Drucker anders reagieren, müssen Sie die gewünschte Zeilenspationierung durch einige Versuche herausfinden.

50: Es werden zwei WINDOWS definiert. WINDOW#0 steht dabei für die Texteingabe zur Verfügung, WINDOW#1 für die Menüeinblendungen.

60: Sprung zum Hauptmenü. Ebenso wie alle anderen Routinen, die mit der direkten Texteingabe nichts zu tun haben, wurde das Hauptmenü an das Programmende gelegt. Dadurch erhöht sich die Ablaufgeschwindigkeit des Programms bei der Texteingabe, weil der Rechner bei Sprungbefehlen (GOTO, GOSUB usw.) intern immer das gesamte Listing nach der angegebenen Zeile durchsucht.

80: Auf dem Bildschirm erscheint die aktuelle Stringnummer, gefolgt von dem Fragezeichen für die Eingabe. Hier können Sie nun entweder Ihren Text oder eine Steuerfunktion (z.B. m oder M für den Sprung zum Hauptmenü) eingeben. Die Texteingaben werden gespeichert; die Steuerfunktionen hingegen belegen keinen Index von T\$(X).

90 bis 160: Hier wird untersucht, ob mit der Eingabe eine Steuerfunktion aufgerufen wurde (vergl. Bedienungshinweise Seite 35). Durch diese Möglichkeiten dürfen Sie natürlich keinen Text eingeben, der mit den Steuercodes identisch ist; d.h. eine Textzeile darf nicht nur aus dem Buchstaben m oder M bestehen, da dies als Sprung zum Hauptmenü interpretiert wird. Erlaubt ist jedoch, Leerzeichen+m oder m. einzugeben. Falls Sie bei Ihren Texten unbedingt Eingaben machen müssen, die mit den Steuercodes identisch sind, ändern Sie am besten diese Codes nach eigenem Geschmack; z.B. #M, *E o.ä.

180 bis 260: Während der Texteingabe lassen sich jederzeit einzelne Strings ändern. Geben Sie dazu statt eines Textes ein e oder E ein. Dadurch gelangen Sie in den Editiermodus. Hier können Sie die gewünschte Stringnummer aufrufen und umändern.

270 bis 310: Mit der Top-Funktion (Eingabe t oder T) gelangen

Sie augenblicklich zum Textanfang. Die aktuelle Stringnummer wird dabei nicht gelöscht.

320 bis 370: Die Bottom-Funktion (Eingabe b oder B) bringt Sie zum Dateiende. Es werden die fünfzehn letzten Textstrings ausgegeben.

380 bis 650: Mit KlarText haben Sie auch die Möglichkeit, einen ganzen Textblock zu bearbeiten. Mit der Eingabe k oder K gelangen Sie zum entsprechenden Unterprogramm. Im WINDOW#1 werden die weiteren Möglichkeiten eingeblendet: K für "Block an andere Stelle kopieren", V für "Block an andere Stelle verschieben" und L für "Block löschen". Sie müssen jeweils die betreffenden Stringnummern eingeben, die den Block umfassen und bestimmen, an welche Stelle der Block kopiert bzw. verschoben werden soll. Handelt es sich dabei nur um einen einzelnen String, sind Anfangs- und Endnummern identisch. Nach der Blockbearbeitung folgt die Abfrage, mit welcher Stringnummer Sie weiterarbeiten wollen. Sie können also z.B. einen Block mit den Stringnummern 11 bis 15 an die Stelle 25 bis 29 verschieben und anschließend bei Nummer 11 normal weiterarbeiten. Im normalen Textmodus läßt sich zudem durch die Eingabe x oder X jederzeit eine neue Stringnummer bestimmen, bei der Sie weiterarbeiten möchten. Dadurch wird das Textverarbeitungsprogramm sehr flexibel.

660 bis 870: Neben der Top- und Bottom-Funktion können Sie im Text beliebig aufwärts oder abwärts rollen. Dabei werden jeweils fünfzehn Textstrings auf dem Bildschirm ausgegeben.

880 bis 910: Ab hier beginnen die Unterprogramme, die vom Hauptmenü aus angesteuert werden können (vergl. Bildschirm 1).

920 bis 980: Im Texteingabemodus erscheint unten auf dem Bildschirm eine Zeilenhilfe. Das Ausrufungszeichen markiert jede fünfte Spalte. Sie können damit Ihre Zeilenbreite leicht bestimmen.

990 bis 1110: Das Unterprogramm zum Abspeichern einer Textdatei. Zum eingegebenen Dateinamen wird der Zusatz .DAT addiert. Das bewirkt, daß die Datei im ASCII-Format gespeichert wird. Je nachdem, welches Speichermedium Sie verwenden (Original Schneider-Laufwerk, ESCON 5 1/4 Zoll-Laufwerk etc.) kann es ggf. zu kleinen Datenübertragungsschwierigkeiten kommen (z.B. Aufspaltung der Textstrings bei Kommaeingaben). Versuchen Sie in diesen Fällen die speziellen Dateikommandos wie LIST# oder WRITE anstelle des PRINT#-Befehls.

1120 bis 1230: Einladen einer Textdatei von Diskette. Nach Eingabe des Dateinamens werden die mit INPUT#n eingelesenen Datensätze dem aktuellen Index von T\$(X) zugewiesen. Um Fehlermeldungen zu vermeiden, muß eine Datei bereits vor dem Abspeichern mit .ende oder .ENDE abgeschlossen werden.

1440 bis 1670: Zum Ausdruck einer Datei stehen Ihnen zahlreiche Optionen zur Verfügung. Mit den Punktcommandos auf Seite 36 können Sie innerhalb des Textes die Darstellungsart des Ausdrucks beeinflussen, die Formularlänge bestimmen usw. In diesem Unterprogramm haben Sie zusätzlich die Möglichkeit, bestimmte Standardformate voreinzustellen. Die Menüabfragen sind dabei so gestaltet, daß die angenommenen Antworten (Ja oder Nein) durch Drücken von RETURN eingegeben werden. Nur beim Abweichen von diesen Standardformaten müssen die Angaben einzeln gemacht werden. Besondere Sorgfalt muß beim Abtippen der Zeilen 1530 bis 1610 herrschen. Die hier untergebrachten String-Manipulationen untersuchen den Text nach Punktcommandos. Zeile 1530 z.B. würde eingedeutscht lauten "Wenn die drei linken Zeichen des aktuellen Strings = .pb lauten, dann drucke den rechten Teil des Strings ab dem vierten Zeichen in Fettschrift." Sobald der String ".ende" oder ".ENDE" eingelesen ist, wird der Druckvorgang beendet.


1680 bis 1930: In diesem Unterprogramm können Sie die voreinstellten Druckercommandos verändern. Wie anfangs bereits erwähnt, wurde das Programm auf den EPSON FX-80 abgestimmt. Die einzelnen Sequenzen müssen bei Verwendung eines anderen Druckers ggf. geändert werden.

1980 bis 2120: Das Hauptmenü. Hier können Sie zwischen neun verschiedenen Punkten wählen oder mit E das Programm beenden.

Drucker-Zeichen

Mit Druckern kann man eine Menge Ärger haben - oder auch Spaß. Denn viele Drucker, die sich an den Schneider-Computer anschließen lassen, verfügen über die Möglichkeit, selbstdefinierte Zeichen auszugeben.


Sie können also ein eigenes Alphabet gestalten oder Sonderzeichen definieren, die Ihnen normalerweise nicht zur Verfügung stehen (z.B. Astrologie-Zeichen oder Buchstaben aus anderen Sprachen). Mit dem folgenden Anwenderprogramm lassen sich Drucker-Zeichen auf dem Bildschirm entwerfen und beliebig verändern. Die Entwürfe können sofort ausgedruckt oder auf Diskette gespeichert werden. Eine Reihe von Sonderfunktionen erleichtern das Arbeiten; z.B. das Füllen oder Löschen ganzer Spalten und Zeilen oder das Invertieren des entworfenen Zeichens.




Das Programm ist auf den Drucker EPSON FX-80 abgestimmt. Bei diesem Modell stehen für eigene Zeichen-Entwürfe 10 mal 9

Punkte zur Verfügung. Dieses Raster wird auf dem Bildschirm abgebildet. Um einen Punkt zu setzen, drücken Sie z.B. H und 0 oder A und 7. Auf diese Weise können Sie einen bereits gesetzten Punkt auch wieder löschen.

Es ist wichtig, daß sich der Rechner im Großschrift-Modus befindet. Nicht erlaubt ist das Setzen von Punkten, die horizontal unmittelbar nebeneinander liegen.



erlaubt



nicht erlaubt

Bedienung des Programms

Eingabe:

Buchstabe + Zahl	Einzelpunkt setzen/löschen
11	gesamtes Raster füllen
22	gesamtes Raster löschen
33	Zeichen invertieren
4 + Buchstabe	Horizontalreihe füllen
5 + Buchstabe	Horizontalreihe löschen
6 + Zahl	Vertikalreihe setzen
7 + Zahl	Vertikalreihe löschen

Ausgabe:

88	Entwurf speichern
99	Dezimalwerte ausdrucken
PP	Entwurf drucken
00	Programm beenden

```

10 'Drucker-Zeichen
20 DIM R(9,7),DW(9)
30 MODE 1
40 WINDOW#0,1,18,1,20:WINDOW#1,1,40,21,25:WINDOW#2,
9,40,1,20
50 'Entwurfsraster
60 PRINT#0," 1 3 5 7 9"
70 PRINT#0," 0 2 4 6 8"
80 PRINT#0," !!!!!!!!!!"
90 PRINT#0,"H-"
100 PRINT#0,"G-"
110 PRINT#0,"F-"
120 PRINT#0,"E-"
130 PRINT#0,"D-"
140 PRINT#0,"C-"
150 PRINT#0,"B-"
160 PRINT#0,"A-"
170 PRINT#2,"11 Raster voll"
180 PRINT#2,"22 Raster leer"
190 PRINT#2,"33 Invertieren":PRINT#2
200 PRINT#2,"4a Horiz. voll"
210 PRINT#2,"5a Horiz. leer":PRINT#2
220 PRINT#2,"6n Vertik. voll"
230 PRINT#2,"7n Vertik. leer":PRINT#2
240 PRINT#2,"88 Speichern"
250 PRINT#2,"99 Werte drucken":PRINT#2
260 PRINT#2,"00 Beenden"
270 PRINT#2,"PP Zeichen drucken"
280 'Eingabe
290 ON ERROR GOTO 510
300 INPUT#1,A$
310 IF A$="11" THEN 600
320 IF A$="22" THEN 680
330 IF A$="33" THEN 760
340 IF LEFT$(A$,1)="4" THEN 840
350 IF LEFT$(A$,1)="5" THEN 930
360 IF LEFT$(A$,1)="6" THEN 1020
370 IF LEFT$(A$,1)="7" THEN 1110
380 IF A$="88" THEN 1200
390 IF A$="99" THEN 1410
400 IF A$="00" THEN 1530
410 IF A$="PP" THEN 1570
420 B=ASC(LEFT$(A$,1))-65
430 R=ASC(RIGHT$(A$,1))-48
440 GOSUB 1830
450 IF R(R,B)=0 OR R(R,B)=32 THEN P=143:GOTO 470
460 IF R(R,B)=143 THEN P=32

```

```

470 PRINT CHR$(7)
480 LOCATE X,Y:PRINT CHR$(P)
490 R(R,B)=P
500 GOTO 290
510 'Fehlerbehebung
520 PRINT CHR$(7)
530 PRINT#1,"Umschalten auf Grossbuchstaben"
540 PRINT#1,"Weiter mit RETURN"
550 PRINT CHR$(7)
560 INPUT#1,B$
570 PRINT#1,CHR$(12)
580 PRINT#1,"Bereit zur Eingabe"
590 GOTO 290
600 'Raster voll
610 P1=4:P2=4
620 FOR F1=0 TO 9:FOR F2=0 TO 7
630 R(F1,F2)=143
640 LOCATE P1+F1,P2+F2:PRINT CHR$(143)
650 PRINT CHR$(7):NEXT F2:NEXT F1
660 P1=0:P2=0
670 GOTO 290
680 'Raster leer
690 P1=4:P2=4
700 FOR F1=0 TO 9:FOR F2=0 TO 7
710 R(F1,F2)=32
720 LOCATE P1+F1,P2+F2:PRINT CHR$(32)
730 PRINT CHR$(7):NEXT F2:NEXT F1
740 P1=0:P2=0
750 GOTO 290
760 'Invertieren
770 FOR R=0 TO 9:FOR B=0 TO 7
780 GOSUB 1830
790 IF R(R,B)=32 OR R(R,B)=0 THEN R(R,B)=143:LOCATE
X,Y:PRINT CHR$(143):GOTO 810
800 IF R(R,B)=143 THEN R(R,B)=32:LOCATE X,Y:PRINT CH
R$(32)
810 PRINT CHR$(7)
820 NEXT B:NEXT R
830 GOTO 290
840 'Horizontal voll
850 B=ASC(RIGHT$(A$,1))-65
860 FOR R=0 TO 9
870 GOSUB 1830
880 R(R,B)=143
890 LOCATE X,Y:PRINT CHR$(143)
900 PRINT CHR$(7)
910 NEXT R

```

```

920 GOTO 290
930 'Horizontal voll
940 B=ASC(RIGHT$(A$,1))-65
950 FOR R=0 TO 9
960 GOSUB 1830
970 R(R,B)=32
980 LOCATE X,Y:PRINT CHR$(32)
990 PRINT CHR$(7)
1000 NEXT R
1010 GOTO 290
1020 'Vertikal voll
1030 R=ASC(RIGHT$(A$,1))-48
1040 FOR B=0 TO 7
1050 GOSUB 1830
1060 R(R,B)=143
1070 LOCATE X,Y:PRINT CHR$(143)
1080 PRINT CHR$(7)
1090 NEXT B
1100 GOTO 290
1110 'Vertikal leer
1120 R=ASC(RIGHT$(A$,1))-48
1130 FOR B=0 TO 7
1140 GOSUB 1830
1150 R(R,B)=32
1160 LOCATE X,Y:PRINT CHR$(32)
1170 PRINT CHR$(7)
1180 NEXT B
1190 GOTO 290
1200 'Zeichen speichern
1210 FOR R=0 TO 9:DW(R)=0:NEXT R
1220 DN$="ZEICHEN."
1230 PRINT#1,"Welches Zeichen soll";
1240 INPUT#1," ersetzt werden";X$
1250 D=ASC(X$)
1260 PRINT#1,CHR$(12)
1270 PRINT#1,"Einen Moment Geduld bitte"
1280 FOR R=0 TO 9
1290 Z=0
1300 FOR B=0 TO 7
1310 IF R(R,B)=143 THEN DW(R)=DW(R)+(2^Z)
1320 Z=Z+1:PRINT CHR$(7)
1330 NEXT B:NEXT R
1340 OPENOUT DN$+X$
1350 PRINT#9,27;38;0;D;D;139
1360 FOR R=0 TO 9:PRINT#9,DW(R);:NEXT R
1370 CLOSEOUT
1380 PRINT CHR$(7)

```

```

1390 PRINT#1,CHR$(12)
1400 GOTO 290
1410 'Dezimalwerte drucken
1420 PRINT#1,CHR$(12)
1430 FOR R=0 TO 9:DW(R)=0:NEXT R
1440 FOR R=0 TO 9
1450 Z=0
1460 FOR B=0 TO 7
1470 IF R(R,B)=143 THEN DW(R)=DW(R)+(2^Z)
1480 Z=Z+1:PRINT CHR$(7)
1490 NEXT B:NEXT R
1500 FOR R=0 TO 9:PRINT#8,DW(R);:NEXT R
1510 PRINT#8
1520 GOTO 290
1530 'Programm beenden
1540 PRINT CHR$(7)
1550 MODE 1
1560 END
1570 'Zeichen drucken
1580 PRINT#1,"Welches Zeichen soll";
1590 INPUT#1," ersetzt werden";X$
1600 D=ASC(X$)
1610 PRINT#1,"Befinden sich die Werte im Speicher"
1620 INPUT#1,"oder auf Diskette (S/D)";Y$
1630 IF Y$="D" THEN 1780
1640 FOR R=0 TO 9:DW(R)=0:NEXT R
1650 FOR R=0 TO 9
1660 Z=0
1670 FOR B=0 TO 7
1680 IF R(R,B)=143 THEN DW(R)=DW(R)+(2^Z)
1690 Z=Z+1:PRINT CHR$(7)
1700 NEXT B:NEXT R
1710 PRINT#8,CHR$(27);"&";CHR$(0);CHR$(D);CHR$(D);CHR$(139);
1720 FOR R=0 TO 9:PRINT#8,CHR$(DW(R));:NEXT R
1730 PRINT#8,CHR$(0)
1740 PRINT#8,CHR$(27);"%";CHR$(1);CHR$(0);X$;CHR$(13)
)
1750 PRINT#8,CHR$(27);"%";CHR$(0);CHR$(0);CHR$(13)
1760 PRINT#1,CHR$(12)
1770 GOTO 290
1780 'Daten auf Diskette
1790 OPENIN "ZEICHEN."+X$
1800 IF EOF THEN 1820
1810 INPUT#9,T:PRINT#8,CHR$(T);:GOTO 1810
1820 GOTO 1740
1830 IF B=0 THEN Y=11

```

```

1840 IF B=1 THEN Y=10
1850 IF B=2 THEN Y=9
1860 IF B=3 THEN Y=8
1870 IF B=4 THEN Y=7
1880 IF B=5 THEN Y=6
1890 IF B=6 THEN Y=5
1900 IF B=7 THEN Y=4
1910 IF R=0 THEN X=4
1920 IF R=1 THEN X=5
1930 IF R=2 THEN X=6
1940 IF R=3 THEN X=7
1950 IF R=4 THEN X=8
1960 IF R=5 THEN X=9
1970 IF R=6 THEN X=10
1980 IF R=7 THEN X=11
1990 IF R=8 THEN X=12
2000 IF R=9 THEN X=13
2010 RETURN

```

20: Für das Entwurfsraster (10 Reihen mit je 8 Bit) wird die Variable R(X,Y) DIMensioniert. Den Elementen von DW(X) werden später die Dezimalwerte des Bitmusters zugewiesen.

60 bis 160: Das Entwurfsraster wird auf dem Bildschirm ausgegeben.

300: Mit INPUT A\$ wartet der Rechner auf die Eingabe von zwei Zeichen. Bei einer Fehleingabe (z.B. 6 und A oder 4 und 7) wird das Programm nicht abgebrochen, sondern durch die ON ERROR GOTO-Anweisung aus Zeile 290 im Unterprogramm 510 fortgesetzt. Sie können dort RETURN drücken und Ihre Eingaben erneuern.

310 bis 410: Hier wird ermittelt, ob eine der Sonderfunktionen aufgerufen wurde.

420 und 430: Die Bitreihe für eigene Zeichendefinitionen ist bei allen Drucker vertikal angeordnet (nicht horizontal wie bei den Zeichen des Computers). Je nach Druckermodell befindet sich das höherwertige Bit (7) oben oder unten. Beim EPSON FX-80 ist es oben; daher beginnt die Bitreihe mit H und endet unten mit A. Um zu ermitteln, welchen Punkt (= welches Bit) Sie gewählt haben, wird in Zeile 420 zunächst von der Eingabe A\$ der ASCII-Wert des Buchstabens ermittelt. Haben Sie z.B. C und 2 gedrückt, so ist B=66-65, also 1 (Bit 1); haben Sie H

und 2 gedrückt, so ist $B=72-65$, also 7 (Bit 7). In Zeile 430 wird ermittelt, welche Spalte Sie gedrückt haben. Um bei dem eben gezeigten Beispiel zu bleiben: Der ASCII-Wert von 2 ist 50. $50-48=2$ (Spalte 2). Zum einwandfreien Betrieb des Programms ist es somit unerlässlich, daß Sie immer zuerst den Buchstaben (A bis H) und dann die Zahl (0 bis 9) eingeben. Nur zum Aufruf der Sonderfunktionen wählen Sie die Kombination, die oben abgebildet ist.

440: Im Unterprogramm 1830 wird ermittelt, an welcher Bildschirmposition (Variablen X,Y) der gewählte Punkt gesetzt bzw. gelöscht werden soll.

450 und 460: Hier wird festgestellt, ob das gewählte Bit gesetzt (ASCII-Wert 143) oder nicht gesetzt ist (ASCII-Wert 0 oder 32).

470: Mit dem Kontrollsignal CHR\$(7) erscheint an der gewählten Bildschirmposition ein Punkt oder ein Leerzeichen.

510 bis 590: Die häufigste Fehlerursache ist in diesem Programm das unterlassene Umschalten auf Großschrift. Daher wird bei Auftreten eines ERRORs (vergl. Zeile 290) hier ein Hinweis darauf gegeben. Der Rechner springt allerdings auch in dieses Unterprogramm, wenn Sie z.B. eine Sonderfunktion falsch anwählen. Drücken Sie in diesem Fall einfach RETURN und geben Sie Ihre Anweisung noch einmal richtig ein.

600 bis 670: Die Befehle zum Ausfüllen des gesamten Rasters mit Punkten. Durch die beiden FOR...NEXT-Schleifen in Zeile 620 werden alle Bits aufgerufen und mit der PRINT-Anweisung aus Zeile 640 als volle Punkte auf dem Bildschirm ausgegeben.

680 bis 750: Die Umkehrung der Anweisungen aus den Zeilen 600 bis 670. Hier werden alle Bits gelöscht und das Bildschirmraster mit Leerzeichen gefüllt.

760 bis 830: Zum Invertieren eines entworfenen Zeichens muß zunächst festgestellt werden, welches Bit gesetzt (ASCII-Wert 143) und welches nicht gesetzt ist (ASCII-Wert 0 oder 32). Dementsprechend werden die Werte ausgetauscht, so daß eine Invertierung entsteht.

840 bis 1190: In diesen Abschnitt springt der Rechner, wenn Sie die Sonderfunktionen zum horizontalen oder vertikalen Füllen bzw. Löschen ganzer Reihen aufrufen. Es wird hier das gleiche Prinzip angewandt wie beim Füllen oder Löschen des

gesamten Entwurfsrasters (s. 600 bis 750).

1210: Vor dem Abspeichern eines Zeichens werden die Elemente der indizierten Variable DW(X) mit Null belegt (es könnten sich noch Werte aus alten Entwürfen darin befinden).

1220: Für jedes entworfene Zeichen wird eine sequentielle Datei eröffnet. Sie können sowohl eine Diskettenstation als auch einen Datenrecorder verwenden. Die Dateinamen beginnen immer mit dem Wort "ZEICHEN". Danach folgt als Extender das Zeichen selbst. Auf diese Weise haben Sie immer eine geordnete Datei, aus der Sie einzelne Zeichen oder alle eigenen Entwürfe nacheinander aufrufen können.

1280 bis 1330: Hier werden die Dezimalwerte des entworfenen Zeichens ausgerechnet. Dank der hohen Taktfrequenz des CPC-Computers (über 3 MHz) benötigt der Rechner nur etwa zwei Sekunden.

1350: In dieser Zeile sind die spezifischen CHR\$-Codes enthalten, die der EPSON-Drucker zur Darstellung frei definierter Zeichen benötigt.

1360: Die Dezimalwerte des Entwurfs werden zur Diskette übertragen.

1410 bis 1520: Falls Sie die Dezimalwerte eines selbst-entworfenen Zeichens ausdrucken wollen, steht Ihnen dieses Unterprogramm zur Verfügung. Hier werden zunächst die Werte ausgerechnet (wie in den Zeilen 1280 bis 1330) und danach ausgedruckt.


1570 bis 1820: Neben der Möglichkeit, einen Entwurf zu speichern oder die Dezimalwerte auszudrucken, können Sie sich auch das Zeichen selbst ausdrucken lassen. In diesem Unterprogramm wird gefragt, ob das Zeichen gedruckt werden soll, das sich augenblicklich im Speicher befindet oder eines, das (auf Diskette) gespeichert ist. Durch den Aufruf eines gespeicherten Zeichens wird der Speicher nicht gelöscht. Sie können also Zeichen entwerfen, mit anderen Entwürfen vergleichen, ggf. Änderungen vornehmen und dann erst den neuen Entwurf abspeichern.


Zeichensatz-Generator

Von selbstdefinieren Druckerzeichen zu selbstdefinierten Bildschirmzeichen ist es nur ein kleiner Schritt. Im Gegensatz zu Computern der älteren Generation, wo man frei gestaltete Zeichen nur mit Hilfe von komplizierten POKE-Anweisungen auf den Bildschirm bringen konnte, hält der Schneider-Computer hierfür den einfachen SYMBOL-Befehl bereit. Damit läßt sich ein gesamtes Alphabet nach eigenem Geschmack gestalten.

Ebenso ist es möglich, den Zeichensatz des Rechners so umzugestalten, daß Spielfiguren oder grafische Symbole entstehen, die man wie jedes andere Textzeichen auf den Bildschirm bringen kann.

In den Programmen "Archiv" und "Terminkalender" wurden bereits mit Hilfe des SYMBOL-Befehls Umlaute definiert. Mit dem folgenden Anwenderprogramm können Sie nun auf bequeme Weise alle möglichen Zeichen entwerfen, sich das Ergebnis ansehen, Veränderungen anbringen und jedes kreierte Zeichen auf Diskette abspeichern.





```

10 'Zeichensatz-Generator
20 DIM R(7,7),DW(96,7)
30 GOTO 1940
40 MODE 1
50 WINDOW#0,1,18,1,20:WINDOW#1,1,40,21,25:WINDOW#2,1
9,40,1,20
60 SYMBOL AFTER 32
70 'Entwurfsraster
80 PRINT#0," 1 3 5 7"
90 PRINT#0," 0 2 4 6"
100 PRINT#0,"  !!!!!!!"
110 PRINT#0,"A-"
120 PRINT#0,"B-"
130 PRINT#0,"C-"
140 PRINT#0,"D-"
150 PRINT#0,"E-"
160 PRINT#0,"F-"
170 PRINT#0,"G-"
180 PRINT#0,"H-"
190 PRINT#2,"11 Raster voll"
200 PRINT#2,"22 Raster leer"
210 PRINT#2,"33 Invertieren":PRINT#2
220 PRINT#2,"4a Horiz. voll"
230 PRINT#2,"5a Horiz. leer":PRINT#2

```

```

240 PRINT#2,"6n Vertik. voll"
250 PRINT#2,"7n Vertik. leer":PRINT#2
260 PRINT#2,"88 Speichern"
270 PRINT#2,"99 Werte drucken":PRINT#2
280 PRINT#2,"00 Beenden"
290 PRINT#2,"ZZ Zeichen darstellen"
300 'Eingabe
310 GOTO 600
320 ON ERROR GOTO 540
330 INPUT#1,A$
340 IF A$="11" THEN 630
350 IF A$="22" THEN 710
360 IF A$="33" THEN 790
370 IF LEFT$(A$,1)="4" THEN 870
380 IF LEFT$(A$,1)="5" THEN 960
390 IF LEFT$(A$,1)="6" THEN 1050
400 IF LEFT$(A$,1)="7" THEN 1140
410 IF A$="88" THEN 1230
420 IF A$="99" THEN 1410
430 IF A$="00" THEN 1560
440 IF A$="ZZ" THEN 1600
450 B=ASC(LEFT$(A$,1))-65
460 R=ASC(RIGHT$(A$,1))-48
470 GOSUB 1770
480 IF R(R,B)=0 OR R(R,B)=32 THEN P=143:GOTO 500
490 IF R(R,B)=143 THEN P=32
500 PRINT CHR$(7);
510 LOCATE X,Y:PRINT CHR$(P)
520 R(R,B)=P
530 GOTO 320
540 'Fehlerbehebung
550 PRINT CHR$(7)
560 PRINT#1,"Umschalten auf Grossbuchstaben"
570 PRINT#1,"Weiter mit RETURN"
580 PRINT CHR$(7);
590 INPUT#1,B$
600 PRINT#1,CHR$(12)
610 PRINT#1,"Bereit zur Eingabe"
620 GOTO 320
630 'Raster voll
640 P1=4:P2=4
650 FOR F1=0 TO 7:FOR F2=0 TO 7
660 R(F1,F2)=143
670 LOCATE P1+F1,P2+F2:PRINT CHR$(143)
680 PRINT CHR$(7):NEXT F2:NEXT F1
690 P1=0:P2=0
700 GOTO 320

```

```

710 'Raster leer
720 P1=4:P2=4
730 FOR F1=0 TO 7:FOR F2=0 TO 7
740 R(F1,F2)=32
750 LOCATE P1+F1,P2+F2:PRINT CHR$(32)
760 PRINT CHR$(7):NEXT F2:NEXT F1
770 P1=0:P2=0
780 GOTO 320
790 'Invertieren
800 FOR R=0 TO 7:FOR B=0 TO 7
810 GOSUB 1770
820 IF R(R,B)=32 OR R(R,B)=0 THEN R(R,B)=143:LOCATE
X,Y:PRINT CHR$(143):GOTO 840
830 IF R(R,B)=143 THEN R(R,B)=32:LOCATE X,Y:PRINT CH
R$(32)
840 PRINT CHR$(7)
850 NEXT B:NEXT R
860 GOTO 320
870 'Horizontal voll
880 B=ASC(RIGHT$(A$,1))-65
890 FOR R=0 TO 7
900 GOSUB 1770
910 R(R,B)=143
920 LOCATE X,Y:PRINT CHR$(143)
930 PRINT CHR$(7)
940 NEXT R
950 GOTO 320
960 'Horizontal vol
970 B=ASC(RIGHT$(A$,1))-65
980 FOR R=0 TO 7
990 GOSUB 1770
1000 R(R,B)=32
1010 LOCATE X,Y:PRINT CHR$(32)
1020 PRINT CHR$(7)
1030 NEXT R
1040 GOTO 320
1050 'Vertikal voll
1060 R=ASC(RIGHT$(A$,1))-48
1070 FOR B=0 TO 7
1080 GOSUB 1770
1090 R(R,B)=143
1100 LOCATE X,Y:PRINT CHR$(143)
1110 PRINT CHR$(7)
1120 NEXT B
1130 GOTO 320
1140 'Vertikal leer
1150 R=ASC(RIGHT$(A$,1))-48

```

```

1160 FOR B=0 TO 7
1170 GOSUB 1770
1180 R(R,B)=32
1190 LOCATE X,Y:PRINT CHR$(32)
1200 PRINT CHR$(7)
1210 NEXT B
1220 GOTO 320
1230 'Zeichen speichern
1240 PRINT#1,"Welches Zeichen soll";
1250 INPUT#1," ersetzt werden";X$
1260 DD=ASC(X$)
1270 FOR R=0 TO 7:DW(DD,R)=0:NEXT R
1280 IF DN$="" THEN INPUT#1,"Zuerst Dateiname defini
eren";DN$:OPENOUT DN$
1290 PRINT#1,CHR$(12)
1300 PRINT#1,"Einen Moment Geduld bitte"
1310 FOR R=0 TO 7
1320 Z=0
1330 FOR B=0 TO 7
1340 IF R(R,B)=143 THEN DW(DD,R)=DW(DD,R)+(2^Z)
1350 Z=Z+1
1360 NEXT B:NEXT R
1370 PRINT#9,DD;
1380 FOR R=0 TO 7:PRINT#9,DW(DD,R);:NEXT R
1390 PRINT CHR$(7)
1400 GOTO 600
1410 'Dezimalwerte einzelner Zeichen drucken
1420 PRINT#1,CHR$(12)
1430 PRINT#1,"Welches Zeichen soll";
1440 INPUT#1," gedruckt werden";X$
1450 DD=ASC(X$)
1460 FOR R=0 TO 7:DW(DD,R)=0:NEXT R
1470 FOR R=0 TO 7
1480 Z=0
1490 FOR B=0 TO 7
1500 IF R(R,B)=143 THEN DW(DD,R)=DW(DD,R)+(2^Z)
1510 Z=Z+1
1520 NEXT B:NEXT R
1530 FOR R=0 TO 7:PRINT#8,DW(DD,R);:NEXT R
1540 PRINT#8
1550 GOTO 320
1560 'Programm beenden
1570 CLOSEOUT
1580 MODE 1
1590 END
1600 'Zeichen darstellen
1610 PRINT#1,CHR$(12)

```

```

1620 PRINT#1,"Welches Zeichen soll"
1630 INPUT#1,"dargestellt werden";X$
1640 DD=ASC(X$)
1650 FOR R=0 TO 7:DW(DD,R)=0:NEXT R
1660 FOR R=0 TO 7
1670 Z=0
1680 FOR B=0 TO 7
1690 IF R(R,B)=143 THEN DW(DD,R)=DW(DD,R)+(2^Z)
1700 Z=Z+1
1710 NEXT B:NEXT R
1720 SYMBOL DD,DW(DD,0),DW(DD,1),DW(DD,2),DW(DD,3),D
W(DD,4),DW(DD,5),DW(DD,6),DW(DD,7)
1730 PRINT#1,CHR$(12)
1740 PRINT#1,X$;" ";X$;" ";X$;" ";X$;
1750 INPUT#1,"Weiter mit RETURN";X$
1760 GOTO 600
1770 IF B=0 THEN Y=4
1780 IF B=1 THEN Y=5
1790 IF B=2 THEN Y=6
1800 IF B=3 THEN Y=7
1810 IF B=4 THEN Y=8
1820 IF B=5 THEN Y=9
1830 IF B=6 THEN Y=10
1840 IF B=7 THEN Y=11
1850 IF R=0 THEN X=4
1860 IF R=1 THEN X=5
1870 IF R=2 THEN X=6
1880 IF R=3 THEN X=7
1890 IF R=4 THEN X=8
1900 IF R=5 THEN X=9
1910 IF R=6 THEN X=10
1920 IF R=7 THEN X=11
1930 RETURN
1940 MODE 2
1950 PRINT"*****
*****"
1960 PRINT"***** Zeichensatz-Generat
or *** Hauptmenü *****"
1970 PRINT"*****
*****"
1980 PRINT:PRINT
1990 PRINT"1. Zeichensatz neu definieren":PRINT
2000 PRINT"2. Zeichensatz einladen":PRINT
2010 PRINT"3. Dateinamen für geplanten Entwurf defin
ieren"
2020 PRINT:PRINT:PRINT
2030 INPUT" Ihre Eingabe bitte:";X$

```

```

2040 IF X$="1" THEN 40
2050 IF X$="2" THEN 2080
2060 IF X$="3" THEN INPUT"Eingabe Dateiname:";DN$:GO
TO 1940
2070 GOTO 1940
2080 `Zeichensatz einladen
2090 INPUT "Eingabe Dateiname";DN$
2100 OPENIN DN$
2110 R=0
2120 IF EOF THEN CLOSEIN:GOTO 2180
2130 INPUT#9,DA
2140 INPUT#9,DB
2150 DW(DA,R)=DB
2160 R=R+1:IF R=8 THEN R=0;GOTO 2120
2170 GOTO 2140
2180 PRINT"Datei eingeladen. Weiter mit RETURN";:INP
UT X$:GOTO 1940

```

20: Für das Entwurfsraster von 8 mal 8 Punkten wird eine entsprechende Feldvariable definiert. Mit DW(96,7) können Sie insgesamt 97 Entwürfe gleichzeitig im Speicher verwalten.

60: Vor der Eingabe von Daten mit dem SYMBOL-Befehl muß durch SYMBOL AFTER festgelegt werden, ab welchem Zeichen eine Umgestaltung vorgenommen werden soll. In diesem Fall ab dem ASCII-Code 32 (Leerzeichen).

70 bis 530: Das Entwurfsraster und die aufgelisteten Sonderfunktionen wie Invertieren des entworfenen Zeichens usw. entsprechen im Wesentlichen denen des Programms "Drucker-Zeichen" (Zeilen 60 bis 500). Auch die Eingabe- und Bedienungsfunktionen (z.B. A0 zum Setzen oder Löschen eines Punktes) sind mit dem Programm "Drucker-Zeichen" identisch.

630 bis 700: Durch drücken von 11 kann das gesamte Entwurfsraster mit Punkten gefüllt werden. Diese Funktion ist nützlich, wenn z.B. invertierte Zeichen entworfen werden, bei denen nur wenige Punkte nicht gesetzt sind.

710 bis 780: Die Eingabe 22 löscht das gesamte Raster. Sie können dann einen neuen Entwurf beginnen.

790 bis 860: Zum Invertieren eines entworfenen Zeichens werden die Werte der indizierten Variable R(X,Y) untersucht. Wenn ein Element den Wert 32 oder 0 hat (kein Punkt gesetzt), soll es den Wert 143 (Grafikzeichen mit vollem Punkt) annehmen und

umgekehrt. Auf diese einfache Weise entsteht die Invertierung eines Entwurfs.

870 bis 1220: Durch entsprechende Eingaben lassen sich horizontal und vertikal einzelne Reihen bzw. Spalten mit Punkten füllen oder gesetzte Punkte wieder löschen. Diese Funktion erleichtert ebenfalls die Zeichengestaltung.

1230 bis 1400: Im Speicher lassen sich bis zu 97 einzelne Entwürfe gleichzeitig verwalten. Es ist aber auch möglich, jederzeit die Daten eines Zeichens auf Diskette oder Kassette zu sichern. Diese Funktion kann unabhängig von einem momentanen Entwurf durchgeführt werden; d.h. das Zeichen, das sich augenblicklich auf dem Bildschirm befindet, wird nicht gelöscht.

1410 bis 1550: Neben dem Abspeichern von Daten für einen Entwurf ist es möglich, diese Daten auch ausdrucken zu lassen. Das kann z.B. sinnvoll sein, wenn Sie die Werte in einem anderen Programm als DATA-Zeilen einbringen wollen.


1600 bis 1760: Mit diesem Unterprogramm kann ein entworfenes Zeichen in Originalgröße dargestellt werden. Bedenken Sie dabei, daß sich hierdurch die Bildschirmdarstellung insgesamt verändert. Wenn Sie z.B. ein "A" zu einem Männlein umdefinieren, werden alle Texte, in denen ein "A" vorkommt, an den betreffenden Stellen ein Männlein aufweisen. Intern verarbeitet der Rechner das Zeichen jedoch weiterhin wie ein "A".

1940 bis 2180: Das Hauptmenü mit verschiedenen Auswahlpunkten. Hier kann auch ein abgespeicherter Zeichensatz eingeladen werden.

Balkendiagramm

Mit dem folgenden Anwenderprogramm haben Sie die Möglichkeit, Umsatzzahlen oder andere statistische Daten grafisch darzustellen. Ein übersichtliches Balkendiagramm sagt mehr aus als nackte Zahlen. Die vorliegende Programmversion erlaubt Eingaben von Umsätzen für einen Zeitraum von zehn Jahren. Natürlich können nach diesem Prinzip auch monatliche Umsätze oder Daten ganz anderer Art eingegeben werden. In diesen Fällen muß das Listings nach eigenen Bedürfnissen umgeschrieben werden. Das Berechnungsprinzip und die grafische Darstellung bleiben dabei erhalten.


Alle eingegebenen Daten können auf Diskette oder Kassette gespeichert oder von dort eingeladen werden. Dadurch ist es möglich, mehrere Statistiken nacheinander aufzurufen und mit-




einander zu vergleichen. Da der Ladevorgang nur ein bis zwei Sekunden dauert, ist rasches und bequemes Arbeiten gewährleistet.

Als weiteren Pluspunkt erlaubt das Programm, ein dargestelltes Balkendiagramm originalgetreu über einen Drucker wiederzugeben (einen Beispielausdruck finden Sie auf Seite 65).

Im Hauptmenü können Sie wählen, ob die Darstellung nur auf dem Bildschirm oder auf Bildschirm und Drucker erfolgen soll. Die vorliegende Version ist auf den EPSON FX-80 abgestimmt. Durch Ändern von nur zwei Zeilen ist es jedoch auch möglich, einen Ausdruck über den STAR radix-10 oder den Schneider-Drucker zu erreichen (s. Listing-Kommentar).



Nach Eingabe der statistischen Daten werden diese Zahlen in grafische Werte umgerechnet und als Balken auf dem Bildschirm dargestellt. Dazu erscheinen die betreffenden Jahres- und Umsatzzahlen. Text und Grafik lassen sich beim Schneider-Computer problemlos mischen. Durch Drücken von RETURN können Sie das Grafikfenster verlassen und zum Hauptmenü zurückkehren.



```

10 'Balkendiagramm
20 MODE 2
30 DIM U(10):PJ=0
40 PRINT"*****
*****"
50 PRINT"***** B A L K E N D I A
G R A M M *****"
60 PRINT"*****
*****"
70 PRINT:PRINT:PRINT
80 PRINT"A Ausgabe nur mit Bildschirm"
90 PRINT"B Ausgabe mit Bildschirm und Drucker"
100 PRINT"S Speichern von ermittelten Daten"
110 PRINT"L Laden von gespeicherten Daten"
120 PRINT"D Neues Diagramm (bei mehrfachem Programm
durchlauf)"
130 PRINT
140 PRINT"E Programm beenden"
150 PRINT:PRINT
160 INPUT"Ihre Eingabe bitte: ";A$
170 IF A$="s" OR A$="S" THEN 670
180 IF A$="d" OR A$="D" THEN RUN
190 IF A$="e" OR A$="E" THEN MODE 1:END
200 IF A$="l" OR A$="L" THEN 770
210 CLS:Y=0
220 PRINT"*****
*****"
230 PRINT"***** D A T E N E I
N G A B E *****"
240 PRINT"*****
*****"
250 PRINT:PRINT
260 INPUT"Wieviele Jahre sollen erfasst werden (1 bi
s 10) ";AJ
270 INPUT"Beginn mit dem Jahr ";BJ
280 PRINT
290 FOR X=BJ TO BJ+AJ-1:PRINT"Eingabe Umsatz aus Jah
r";X;": ";:INPUT U(Y):Y=Y+1:NEXT X
300 'Balkendiagramm darstellen
310 CLS:Y=0:P=2
320 FOR X=1 TO AJ:LOCATE 1,P:PRINT BJ:P=P+2:BJ=BJ+1:
NEXT X
330 PLOT 50,50:DRAW 50,399:PLOT 50,50:DRAW 635,50
340 P=100
350 FOR X=0 TO 11
360 PLOT P,50:DRAW P,399
370 P=P+48.5

```

```

380 NEXT X
390 P=6:Y=0
400 FOR X=0 TO 11:LOCATE P,24:PRINT Y:P=P+6:Y=Y+10:N
EXT X
410 Y=0:P=382
420 FOR X=1 TO AJ
430 F=INT(U(Y)/1000/2)
440 P2=INT(U(Y)/1000+50)+(8*F)-(F/4)
450 PLOT 50,P:DRAW P2,P:DRAW P2,P-14:DRAW 50,P-14
460 FOR XX=13 TO 1 STEP-1:PLOT 50,P-XX:DRAW P2,P-XX:
NEXT XX
470 Y=Y+1:P=P-32
480 NEXT X
490 IF A$="b" OR A$="B" THEN GOSUB 520
500 LOCATE 7,25:INPUT"Weiter mit RETURN";A$
510 CLS:GOTO 40
520 'Hardcopy-Routine auf EPSON FX-80
530 DEFINT A-Z
540 ORIGIN 0,0
550 PRINT#8,CHR$(27);"A";CHR$(2);
560 FOR I=399 TO 0 STEP-4
570 PRINT#8,CHR$(27);"*";CHR$(4);CHR$(127);CHR$(2);
580 FOR X1=0 TO 639
590 FOR X2=4 TO 0 STEP-1
600 IF TEST(X1,I-X3)>0 THEN N=N+2^X2
610 X3=X3+1:NEXT X2
620 PRINT#8,CHR$(N);
630 N=0:X3=0:NEXT X1:PRINT#8
640 NEXT I
650 PRINT#8,CHR$(27);"A";CHR$(6)
660 RETURN
670 'Bildschirmdaten abspeichern
680 PRINT:PRINT
690 INPUT"Eingabe Dateiname ";DN$
700 PRINT:PRINT"Bildschirmdaten werden gespeichert.
Bitte warten."
710 OPENOUT DN$+".DAT"
720 BJ=BJ-AJ
730 PRINT#9,AJ:PRINT#9,BJ
740 FOR X=0 TO AJ-1:PRINT#9,U(X):NEXT X
750 CLOSEOUT
760 CLS:GOTO 40
770 'Bildschirmdaten einladen
780 PRINT:PRINT
790 INPUT"Eingabe Dateiname ";DN$
800 PRINT:PRINT"Bildschirmdaten werden geladen. Bitt
e warten."

```

```

810 OPENIN DN$+".DAT"
820 INPUT#9,AJ:INPUT#9,BJ
830 FOR X=0 TO AJ-1:INPUT#9,U(X):NEXT X
840 CLOSEIN
850 GOTO 300

```

30: U(x) ist die indizierte Variable für die Umsatzzahlen. Aus Platzgründen lassen sich nur zehn Jahre erfassen. Falls Sie jedoch auf Zwischenräume verzichten wollen, können Sie auch zwanzig Jahre darstellen. Die nachfolgenden Zählschleifen müssen dann entsprechend erhöht werden.

40 bis 200: Das Hauptmenü wird auf dem Bildschirm dargestellt. Hier können Sie zwischen den beiden Darstellungsarten (Bildschirm und Drucker) wählen und ggf. gespeicherte Daten einladen. Zum Abspeichern von statistischen Werten müssen diese Werte zunächst (nach Drücken von "A") eingegeben werden. Durch die Rückkehr ins Hauptmenü gehen keine Daten verloren.

210 bis 290: Hier werden die statistischen Werte erfaßt. Durch die FOR...NEXT-Schleife in Zeile 290 nimmt die indizierte Variable U(x) die Umsatzzahlen jedes gewünschten Jahres an.

320: Hier werden das Startjahr und alle darauffolgend erfaßten Jahre untereinander gePRINTet. Durch P=P+2 erhöht sich die Zeilenangabe für den LOCATE-Befehl. Falls Sie bis zu zwanzig Jahre erfassen wollen, muß hier P=P+1 stehen.

330: Für das Balkendiagramm werden die Koordinatenlinien gePLOTtet (vergl. Darstellung Seite 65).

350 bis 380: Hier werden die Unterteilungslinien gePLOTtet, die als Orientierungshilfe für die Balken dienen. Insgesamt erscheinen zwölf senkrechte Linien im Abstand von 48.5 Bildpunkten. Dieser Abstand mußte gewählt werden, damit die Linien mit den daruntergePRINTeten Zahlen übereinstimmen.

400: Unter den Koordinaten- bzw. Unterteilungslinien stehen zur Orientierung die Umsatzzahlen (0, 10, 20, 30, usw.). Gemeint sind hier jeweils "Tausend". Falls Sie das Programm zur grafischen Darstellung anderer statistischer Werte benutzen, können Sie anstelle der Zahlen auch andere Zeichen PRINTen.

420 bis 480: Hier werden die eingegebenen Werte in grafische Daten umgerechnet und mit den PLOT...DRAW-Anweisungen dargestellt.

490: Falls im Hauptmenü die Darstellung auf Bildschirm und Drucker gewählt wurde (Eingabe "B"), springt der Rechner ins Unterprogramm 520, wo eine Hardcopy vom Bildschirm angefertigt wird. Leider beansprucht der Ausdruck relativ lange Zeit (bei einer Auflösung von 600 mal 400 Punkten müssen 256.000 Bildpunkte abgefragt werden).

500: Nach der Darstellung der Balken und des evtl. Ausdrucks erscheint in der untersten Zeile der Hinweis "Weiter mit RETURN". Sie können sich also die Grafik so lange ansehen, wie Sie wollen und dann zum Hauptmenü zurückkehren.

530: Da die Hardcopy recht lange dauert, soll der Rechner durch DEFINT A-Z alle Variablen als ganzzahlig behandeln. Dies reduziert die Rechendauer ein wenig.

550: Hier wird der Zeilenabstand für die Bit-Image-Grafik (Einzelpunktgrafik) des Druckers festgelegt. Falls Ihr Drucker kein doppeltes Linefeed (doppelter Zeilenvorschub) leistet, muß anstelle von CHR\$(2) ein CHR\$(4) eingegeben werden.

570: Die Anweisung zur Aktivierung von Bit-Image-Grafik für den EPSON FX-80. Der STAR radix-10 und fast alle anderen Drucker benutzen hierfür den Code CHR\$(27);"K";CHR\$(127);CHR\$(2). Dabei sind die letzten beiden Zahlen flexibel. Sie richten sich nach der Anzahl der Datenbit, die für die ausdruckende Zeile an den Drucker gesendet werden.

580 bis 640: Hier wird ermittelt, ob ein Bildschirmpunkt gesetzt ist oder nicht. Dementsprechend werden auch die Bits für die Einzelpunktgrafik des Druckers gesetzt.

650: Vor der Rückkehr aus dem Unterprogramm wird der Zeilenabstand des Druckers wieder normalisiert. Der Wert CHR\$(6) gilt nur für Drucker, die ein doppeltes Linefeed ausführen. Andernfalls muß hier CHR\$(12) eingegeben werden.

670 bis 700: Nach der Rückkehr aus dem Grafikfenster ins Hauptmenü können Sie die aktuellen Daten auf Diskette oder Kassette abspeichern. Der Rechner fragt hier nach dem Dateinamen.

710: Sie können den Dateinamen ohne Extender eingeben. Der Zusatz .DAT (als Erkennung für Daten, die im ASCII-Format gespeichert sind), wird vom Rechner selbstständig hinzugefügt.

730 und 740: Hier werden die Anzahl der Jahre (AJ), der Jah-

resbeginn (BJ) sowie die Umsatzzahlen der indizierten Variable $U(x)$ an die Diskette bzw. Kassette übertragen.


770 bis 850: Mit diesem Unterprogramm lassen sich gespeicherte Daten wieder einlesen. Sie werden danach als Balkendiagramm dargestellt.

Angebots-Plotter

Es gibt auf grafischem Gebiet Anwendungen, für die ein normaler Matrix-Nadeldrucker nicht geeignet ist. Mit einem Plotter hingegen lassen sich viele grafische Aufgaben einfacher und professioneller lösen. Er verfügt über SteuerCodes, mit denen man Kreise ziehen, Quadrate zeichnen oder Flächen schraffieren kann. Das sind Funktionen, die sich mit der Punktmatrix eines Druckers nur mühsam verwirklichen lassen.


Das folgende Programm erstellt ein Preis-Angebot für einen Kunden mit dazu passender Skizze. Darüber hinaus rechnet es (nach Eingabe der Stückzahl und des Einzelpreises) automatisch den Gesamtbetrag aus und addiert die Mehrwertsteuer.

In der vorliegenden Version wird ein Angebot für ein Zwei-Personen-Zelt geplottet, bei dem Sie den Zeltyp und die Abmessungen eingeben können. Die Bezeichnung und die Zahlen werden an den betreffenden Stellen eingesetzt. Nach diesem Prinzip können Sie sich ein Anwenderprogramm für Ihren persönlichen



Einsatz erstellen; denn es ist dem Plotter gleichgültig, ob er Zelte, Fensterscheiben, Grabsteine, Türrahmen oder die Löcher für Schweizer Käse zeichnet.

Das Programm ist auf den Plotter X100S von adcomp abgestimmt, da sich dieses Gerät ohne Probleme an jeden Schneider-Computer anschließen läßt und über normale PRINT#-Anweisungen gesteuert wird.



Nachdem Sie im Hauptmenü Namen, Anschrift, Datum und Angebotsnummer eingegeben haben, erscheinen die Abfragen für den Zeltyp, die Abmessungen und Stückzahl. Umlaute können Sie genau so eingeben, wie im Archivprogramm Seite 15 gezeigt. Das Plotten dauert anschließend etwa 30 Sekunden.


Stephan Ammann
Pfadfinderzubehör
Postfach 1777
CH-4010 Basel

12. Mai 1985

ANGEBOT
Nr. 0815/007-A

Sehr geehrter Interessent!

wir freuen uns über Ihre Anfrage und können dazu folgendes
Angebot unterbreiten: Zelttyp Campman wie Skizze



Süczzahl:	4	Einzelpreis:	DM 245.5
Gesamtpreis	DM 982		
14% MwSt	DM 137.48		
	DM 1119.48		

(Beispiel eines geplotteten Angebots)

```

10 'Angebots-Plotter
20 MODE 2:WINDOW#0,1,80,1,4:WINDOW#1,1,80,5,20
25 GOSUB 10000
30 PRINT"*****
*****"
32 PRINT"***** A N G E B O T S - P
L O T T E R *****"
34 PRINT"*****
*****"
40 INPUT#1,"Eingabe Vor- und Zuname: ";N$
50 INPUT#1,"Eingabe Zusatz      : ";ZU$
60 INPUT#1,"Eingabe Straße     : ";S$
70 INPUT#1,"Eingabe Wohnort    : ";O$
75 PRINT#1
80 INPUT#1,"Eingabe Datum      : ";D$
90 INPUT#1,"Angebots-Nummer    : ";AN$
100 PRINT#1,CHR$(12)
110 INPUT#1,"Eingabe des Zeltyps : ";ZT$
120 INPUT#1,"Außenmaß lange Seite : ";AL$
130 INPUT#1,"Innenmaß lange Seite : ";IL$
140 INPUT#1,"Außenmaß breite Seite : ";AB$
150 INPUT#1,"Innenmaß breite Seite : ";IB$
155 PRINT#1
160 INPUT#1,"Eingabe Stückzahl   : ";SZ
170 INPUT#1,"Eingabe Einzelpreis : ";EP
180 PRINT#1:PRINT#1:PRINT#1,"Einen Moment Geduld bit
te"
190 'Plotter ansteuern
200 PRINT#8,"C0"
210 PRINT#8,CHR$(17)
220 PRINT#8,N$:PRINT#8
230 PRINT#8,ZU$:PRINT#8
240 PRINT#8,S$:PRINT#8
250 PRINT#8,O$:FOR X=1 TO 4:PRINT#8:NEXT X
260 PRINT#8,D$:PRINT#8:PRINT#8,CHR$(18):PRINT#8,"S6"
:PRINT#8,"C3":PRINT#8,"PANGEBOT"
270 PRINT#8,"S4":PRINT#8,CHR$(17):PRINT#8,"Nr. ";AN$
280 PRINT#8,CHR$(18):PRINT#8,"C0":PRINT#8,CHR$(17)
290 PRINT#8,"Sehr geehrter Interessent!":PRINT#8
295 PRINT#8,"wir freuen uns über Ihre Anfrage und kö
nnen dazu folgendes"
300 PRINT#8,"Angebot unterbreiten:  Zeltyp ";ZT$;"
wie Skizze"
310 PRINT#8
320 'Zelt plotten
330 PRINT#8,CHR$(18):PRINT#8,"M400,600"
340 PRINT#8,"D400,600,400,900,800,900,900,750,800,60

```

0,400,600"
350 PRINT#8,"M380,580":PRINT#8,"C3":PRINT#8,"L1":PRI
NT#8,"B7"
360 PRINT#8,"D380,580,350,750,380,920,820,920,1050,7
50,820,580,380,580"
370 PRINT#8,"M350,750"
380 PRINT#8,"D350,750,1050,750"
390 PRINT#8,"M800,600"
400 PRINT#8,"D800,600,800,900"
410 PRINT#8,"C0":PRINT#8,"L0"
420 PRINT#8,"M190,580"
430 PRINT#8,"D190,580,300,580"
440 PRINT#8,"M200,570"
450 PRINT#8,"D200,570,200,710":PRINT#8,"R0,80":PRINT
#8,"D200,930"
460 PRINT#8,"M190,920"
470 PRINT#8,"D190,920,300,920"
480 PRINT#8,"M210,600"
490 PRINT#8,"D210,600,300,600"
500 PRINT#8,"M220,590"
510 PRINT#8,"D220,590,220,710":PRINT#8,"R0,80":PRINT
#8,"D220,910"
520 PRINT#8,"M210,900"
530 PRINT#8,"D210,900,300,900"
540 PRINT#8,"M350,1000"
550 PRINT#8,"D350,1000,350,1110"
560 PRINT#8,"M340,1100"
570 PRINT#8,"D340,1100,560,1100":PRINT#8,"R80,0":PRI
NT#8,"D1060,1100"
580 PRINT#8,"M1050,1000"
590 PRINT#8,"D1050,1000,1050,1110"
600 PRINT#8,"M400,1000"
610 PRINT#8,"D400,1000,400,1090"
620 PRINT#8,"M390,1080"
630 PRINT#8,"D390,1080,560,1080":PRINT#8,"R80,0":PRI
NT#8,"D910,1080"
640 PRINT#8,"M900,1000"
650 PRINT#8,"D900,1000,900,1090"
660 PRINT#8,"M565,1095"
670 PRINT#8,"S3"
680 PRINT#8,"P"+AL\$
690 PRINT#8,"M565,1070"
700 PRINT#8,"P"+IL\$
710 PRINT#8,"M205,715"
720 PRINT#8,"Q3"
730 PRINT#8,"P"+AB\$
740 PRINT#8,"M230,715"

```

750 PRINT#8,"P"+IB$
760 PRINT#8,"Q0":PRINT#8,"S4"
770 PRINT#8,"M0,500"
780 PRINT#8,CHR$(17)
790 PRINT#8,"Süczahl:      ";SZ;"      Einzelpreis:
   DM ";EP
800 PRINT#8
810 PRINT#8,"Gesamtpreis      DM ";EP*SZ
820 PRINT#8
830 PRINT#8,"14% MwSt          DM ";0.14*(EP*SZ)
840 PRINT#8,"                -----"
850 PRINT#8,"                DM ";(EP*SZ)+0.14*(EP*SZ)
Z)
860 PRINT#8,CHR$(18):PRINT#8,"H"
10000 SYMBOL AFTER 91
10010 SYMBOL 91,102,24,60,102,126,102,102,0
10020 SYMBOL 92,195,60,102,102,102,102,60,0
10030 SYMBOL 93,102,0,102,102,102,102,60,0
10040 SYMBOL 123,108,0,120,12,124,204,118,0
10050 SYMBOL 124,0,102,0,60,102,102,60,0
10060 SYMBOL 125,0,102,0,102,102,102,62,0
10070 SYMBOL 126,60,102,102,124,102,102,108,224
10080 RETURN

```

200 und 210: Beim Plotter X100S gibt es einen Print-Modus und einen Grafik-Modus. Im Print-Modus können Sie normale PRINT#-Anweisungen geben; z.B. PRINT#8,"ABCDEFGF" oder PRINT#8,A\$. Im Grafik-Modus hingegen werden PRINT#-Anweisungen für Steuerfunktionen benutzt; z.B. PRINT#8,"MO,10" bedeutet "Move (bewege) dich zur Position 0,10". In dieser Form gibt es zahlreiche Befehle. Durch den Code CHR\$(17) schaltet man in den Print-Modus, durch CHR\$(18) in den Grafik-Modus. Der X100S arbeitet mit vier Farben. Mit der Anweisung "C0" (Color 0) wird in diesem Fall der Schwarzstift aufgerufen (falls Sie in der Halterung einen Grünstift haben, erscheint natürlich diese Farbe).

220 bis 250: Durch die Umschaltung in den Print-Modus (Zeile 210) können hier die Anschrift und das Datum geplottet werden.

260: Die Anweisung "S6" ruft im Grafik-Modus die Schriftgröße 6 auf (normal 4). Danach wird mit "C3" der Rotstift gewählt und das Wort "ANGEBOT" geschrieben.

270: Die normale Schriftgröße "S4" wird wieder gewählt und die Angebotsnummer geplottet.

280 bis 320: Es folgt ein Angebots-Text, der von Ihnen beliebig geändert werden kann.

330: Ab hier wird die Skizze des Zeltes geplottet. Die Ausgangsposition ist der Punkt 400,600 (es können die Punkte von 0,0 bis 1920,2032 angesprochen werden). Falls Sie in Ihrer eigenen Programmversion eine längere Textpassage einbauen, müssen Sie die Ausgangsposition des Plotters entsprechend verlegen, damit die Zeichnung nicht in den Text hineinreicht.

340: Der Innenraum des Zeltes entsteht. "D" bedeutet "Draw" (ziehe eine Linie). Es können beliebig viele x/y-Koordinaten folgen.

350: Für das Außenzelt wird die Ausgangsposition "M380,580" angefahren, mit "C3" die Farbe Rot aufgerufen, durch "L1" die Funktion "gestrichelte Linie" gewählt und mit "B7" die Strichlänge festgelegt (Werte von 1 bis 127 sind erlaubt).

360 bis 400: Die Draw-Anweisung für das Außenzelt.

410: Um die Maßlinien (in Schwarz mit "C0") zu zeichnen, muß der Plotter wieder auf "normale Linie" geschaltet werden. Dazu dient die Anweisung "L0".

420 bis 650: Das Plotten der Maßlinien ist reine Fleißarbeit. Wenn Sie eine andere Skizze anfertigen, werden sich diese Anweisungen natürlich von den hier gezeigten unterscheiden. Das Prinzip ist jedoch immer das gleiche: Zuerst den Stift zur gewünschten Position fahren, dann mit der Draw-Anweisung Linien ziehen. Eine angenehme Hilfe ist dabei der Befehl "Rx,y", der "relative move" bedeutet, d.h. der Stift bewegt sich von der momentanen Position aus um die Anzahl der Punkte, die bei x,y angegeben sind. Auf diese Weise sind z.B. hier die Zwischenräume in den Maßlinien entstanden, in die nachfolgend die Maße eingetragen werden.

660: Der Stift wird zu der Position bewegt, an der die erste Zahl stehen soll (Außenmaß lang).

670: Die Maßzahlen erscheinen in Schriftgröße 3.

680: Um Zahlen oder Buchstaben schreiben zu können, muß nicht zwangsläufig in den Print-Modus umgeschaltet werden. Mit der

Anweisung "P..." können Textzeichen auch im Grafik-Modus geplottet werden.

690 bis 750: Zuerst werden die beiden Längenmaße geplottet. Um die Breitenmaße um 90 Grad gedreht darzustellen, muß die Anweisung "Q3" gegeben werden; das bedeutet "Schreibrichtung 3" (normal = 0). Mit den Parametern 1 und 2 drehen Sie die Zeichen um 270 bzw. 180 Grad.

760: Zum Plotten des Abschlußtextes und dem automatisch berechneten Gesamtpreis mit Mehrwertsteuer wird wieder normale Schreibrichtung und Schriftgröße 4 gewählt.

860: Die Anweisung "H" bringt den Plotter in die Grundstellung (home). Wird dieser Befehl unterlassen, treten bei einem weiteren Programmdurchlauf Datenübertragungsfehler auf, weil der Plotter über seine Formatbegrenzung hinaus arbeiten soll.

10000 bis 10080: Daten für die Umlaute.

Key-Maker

Das folgende Listing ist kein direktes Anwenderprogramm, sondern eher ein Utility. Mit ihm werden alle Tasten des Zahlenblocks mit BASIC-Befehlen belegt, die während des Programmierens sehr häufig vorkommen; z.B. "list", "auto", "print" usw. Sie können das Utility vor jeder Programmierarbeit einladen und nach dem Starten löschen. Die Funktionstasten bleiben bis zum Ausschalten des Rechners erhalten.

Das folgende Bild zeigt die Belegung der Tasten des Zahlenblocks, wie sie durch das Programm erreicht wird. Sie können natürlich auch eine andere Belegung wählen.

print" (7)	print#8, (8)	print#9, (9)
renum (4)	save" (5)	load" (6)
list (1)	auto (2)	delete (3)
list+RET (0)	list#8 (.)	run+RET (enter)

```

10 '*****
20 '***** K E Y - M A K E R *****
30 '*****
40 '
50 'Zahlentaste 0 = 128
60 KEY 128,"list"+CHR$(13)
70 'Zahlentaste 1 = 129
80 KEY 129,"list "
90 'Zahlentaste 2 = 130
100 KEY 130,"auto "
110 'Zahlentaste 3 = 131
120 KEY 131,"delete "
130 'Zahlentaste 4 = 132
140 KEY 132,"renum "
150 'Zahlentaste 5 = 133
160 KEY 133,"save "+CHR$(34)
170 'Zahlentaste 6 = 134
180 KEY 134,"load "+CHR$(34)
190 'Zahlentaste 7 = 135
200 KEY 135,"print"+CHR$(34)
210 'Zahlentaste 8 = 136
220 KEY 136,"print#8,"
230 'Zahlentaste 9 = 137
240 KEY 137,"print#9,"
250 'Dezimalpunkt-Taste im Zahlenblock = 138
260 KEY 138,"list#8"
270 'Enter-Taste im Zahlenblock = 139
280 KEY 139,"run"+CHR$(13)


```

Die Zuordnung der BASIC-Befehle für die einzelnen Tasten wurde so weit wie möglich logisch gewählt. So sind die Zahlentasten 8 mit "print#8" (Drucker) und 9 mit "print#9" (Diskette oder Kassette) belegt. Alle List-Statements befinden sich im unteren Dreieck des Tastenblocks; 0="list" mit RETURN, 1="list" ohne RETURN (zum Eintragen von einzelnen Zeilennummern) und der Dezimalpunkt="list#8" (Drucker). Der Zusatz chr\$(13) bewirkt beim Drücken der Taste ein RETURN, mit chr\$(34) erscheinen Anführungsstriche hinter der Anweisung (z.B. print"). Falls Sie häufiger Stringvariablen ausdrücken, für die keine Anführungsstriche benötigt werden, so können Sie den Zusatz chr\$(34) fortlassen.

Adressen-Verwaltung

Falls Sie mit Ihrem CPC-Computer eine komfortable Adressen-Verwaltung aufbauen wollen, ist das folgende Anwenderprogramm genau das richtige für Sie. Mit ihm können Sie Adressen erfassen, abspeichern, einlesen, ändern, selektieren und mit anderen Dateien vermischen. Zur Ausgabe über den Drucker bieten sich verschiedene Möglichkeiten: einzelne Adressen drucken, gesamte Datei drucken, alphabetisch drucken, selektiv drucken (z.B. nach dem Suchwort "Mahnung") oder Ausdruck der Adressen als Briefkopf mit eventuell nachfolgendem Briefftext. Sie können also auch Rundschreiben mit diesem Programm verfassen und auf Endlospapier ausdrucken lassen.



Bei so viel Komfort müssen Sie allerdings einen Nachteil in Kauf nehmen: Die relativ geringe Speicherkapazität erlaubt nur die Aufnahme einer begrenzten Zahl von Adressen. In diesem Fall wurde das Programm für 80 Datenblöcke eingerichtet. Falls Sie also sehr viele Adressen erfassen wollen, müssen Sie



mehrere Dateien erstellen und z.B. bei einem Rundschreiben nacheinander einlesen. Dadurch ist das alphabetische Ordnen

natürlich problematisch. Andere Funktionen, wie das selektive Ausdrucken können jedoch auch mit der Mehr-Dateien-Verwaltung durchgeführt werden.

Auf dem ersten Bildschirm erscheint das Hauptmenü. Von hier wird in verschiedene Unterprogramme verzweigt.



Ein Datenblock besteht aus Name, Zusatz (z.B. Ing.-Büro), Straße, Ort, Telefon und Bemerkung (z.B. Geburtstag oder Mahnung wegen...). Durch eine Zählvariable können Sie 80 Datenblöcke eingeben. Nach jedem Eintrag erscheint im unteren Bildschirm die Frage, ob Sie weiterarbeiten oder zum Hauptmenü zurückkehren wollen.

Zum Ausdrucken stehen Ihnen mehrere Möglichkeiten zur Verfügung (s. Kommentar zum Listing).

```

10 'Adressenverwaltung
20 GOSUB 10000
30 DIM N$(80),Z$(80),S$(80),O$(80),T$(80),B$(80)
40 MODE 2:WINDOW#0,1,80,1,4:WINDOW#1,1,80,5,20:WINDO
W#2,1,80,21,25
50 PRINT"*****
*****"
60 PRINT"***** ADRESSENVERWALTUNG *
** HAUPTMENÜ *****"
70 PRINT"*****
*****"
80 PRINT#1,"1 Adressen eingeben           5
Adressen selektieren"
90 PRINT#1,"2 Adressen abspeichern       6
Neue Adressendatei"
100 PRINT#1,"3 Adressen einladen        7
Adressen lesen"
110 PRINT#1,"4 Adressen ausdrucken      8
Programm beenden"
120 PRINT#2,"Ihre Eingabe bitte: ";
130 A$=INKEY$:IF A$="" THEN 130
140 IF A$="1" THEN 280
150 IF A$="2" THEN 460
160 IF A$="3" THEN 610
170 IF A$="4" THEN 1610
180 IF A$="5" THEN 1310
190 IF A$="6" THEN RUN
200 IF A$="7" THEN ZX=0:GOSUB 5000:GOTO 760
210 IF A$="8" THEN MODE 2:END
220 GOTO 130
280 'Adressen eingeben
285 GOSUB 5000
290 PRINT"*****
*****"

```

```

300 PRINT"***** ADRESSENVERWALTUNG
*** EINGABE *****"
310 PRINT"*****
*****"
320 INPUT#1,"Vor- und Zuname : ";N$(Z)
330 INPUT#1,"Zusatz : ";Z$(Z)
340 INPUT#1,"Straße : ";S$(Z)
350 INPUT#1,"Wohnort : ";O$(Z)
360 INPUT#1,"Telefon : ";T$(Z)
370 INPUT#1,"Bemerkungen : ";B$(Z)
380 PRINT#2,"Weiter mit <W> oder mit <M> zum Menü"
390 PRINT#2,"Ihre Eingabe :";
400 A$=INKEY$:IF A$="" THEN 400
410 Z=Z+1
420 IF A$="M" OR A$="m" THEN GOSUB 5000:GOTO 50
430 IF A$="W" OR A$="w" AND Z<81 THEN PRINT#1,CHR$(1
2):PRINT#2,CHR$(12):GOTO 320
440 IF Z>80 THEN INPUT#2,"Datei ist voll. Weiter mit
RETURN ";A$:GOSUB 5000:GOTO 50
450 Z=Z-1:GOTO 400
460 'Adressen abspeichern
470 PRINT#2,CHR$(12):INPUT#2,"Eingabe Dateiname: ";D
N$
480 PRINT#2,CHR$(12):PRINT#2,"Einen Moment Geduld bi
tte. Datei ";DN$;" wird gespeichert"
490 ZX=0
500 OPENOUT DN$+".DAT"
510 PRINT#9,N$(ZX)
520 PRINT#9,Z$(ZX)
530 PRINT#9,S$(ZX)
540 PRINT#9,O$(ZX)
550 PRINT#9,T$(ZX)
560 PRINT#9,B$(ZX)
570 ZX=ZX+1
580 IF N$(ZX)<>"" THEN 510
590 CLOSEOUT
600 PRINT#2,CHR$(12):PRINT#2,"Datei ";DN$;" gesicher
t. Ihre Eingabe bitte:":GOTO 130
610 'Adressen einladen
620 PRINT#2,CHR$(12):INPUT#2,"Eingabe Dateiname: ";D
N$
630 INPUT#2,"Soll der momentane Speicherinhalt übers
chrieben werden (J/N) ";A$
640 IF A$="j" OR A$="J" THEN ZX=0
650 IF A$="n" OR A$="N" THEN ZX=Z
660 OPENIN DN$+".DAT"
670 IF EOF THEN 710

```

```

680 INPUT#9,N$(ZX),Z$(ZX),S$(ZX),O$(ZX),T$(ZX),B$(ZX)
)
690 ZX=ZX+1
700 GOTO 670
710 CLOSEIN
720 Z=ZX+1
730 PRINT#2,CHR$(12):INPUT#2,"Eingeladene Adressen l
esen (J/N) ";A$
740 IF A$="j" OR A$="J" THEN PRINT#1,CHR$(12):PRINT#
2,CHR$(12):ZX=0:GOTO 760
750 GOSUB 5000:GOTO 50
760 'Adressen lesen
770 PRINT#1,N$(ZX)
780 PRINT#1,Z$(ZX)
790 PRINT#1,S$(ZX)
800 PRINT#1,O$(ZX)
810 PRINT#1
820 PRINT#1,T$(ZX)
830 PRINT#1,B$(ZX)
840 PRINT#2,CHR$(12):PRINT#2,"Weiterlesen = RETURN
Ändern = A Drucken = D Menü = M"
880 INPUT#2,"Ihre Eingabe bitte :";A$
890 IF A$="a" OR A$="A" THEN 940
900 IF A$="d" OR A$="D" THEN 1110
910 IF A$="m" OR A$="M" THEN GOSUB 5000:GOTO 50
920 ZX=ZX+1:IF N$(ZX-1)="" THEN PRINT#2,CHR$(12):INP
UT#2,"Dateiende erreicht. Weiter mit RETURN ";A$:GOS
UB 5000:GOTO 50
930 PRINT#1:PRINT#1:GOTO 770
940 'Einzelne Adressen ändern
950 PRINT#1,CHR$(12):PRINT#2,CHR$(12):PRINT#1,"*** Ä
nderungsmodus ***":PRINT#1
960 PRINT#1,N$(ZX)
970 PRINT#1,Z$(ZX)
980 PRINT#1,S$(ZX)
990 PRINT#1,O$(ZX)
1000 PRINT#1
1010 PRINT#1,T$(ZX)
1020 PRINT#1,B$(ZX)
1030 PRINT"*****
*****"
1040 INPUT#1,"Vor- und Zuname : ";N$(ZX)
1050 INPUT#1,"Zusatz : ";Z$(ZX)
1060 INPUT#1,"Straße : ";S$(ZX)
1070 INPUT#1,"Wohnort : ";O$(ZX)
1080 INPUT#1,"Telefon : ";T$(ZX)
1090 INPUT#1,"Bemerkungen : ";B$(ZX)

```

```

1100 ZX=ZX+1:A$="":PRINT#2,CHR$(12):GOTO 840
1110 'Einzelne Adressen ausdrucken
1120 PRINT#8,N$(ZX)
1130 PRINT#8,Z$(ZX)
1140 PRINT#8,S$(ZX)
1150 PRINT#8,O$(ZX)
1160 PRINT#8:PRINT#8,T$(ZX)
1170 PRINT#8,B$(ZX)
1180 PRINT#8
1190 PRINT#2,CHR$(12):GOTO 840
1200 'Adressendatei chronologisch drucken
1210 ZY=0
1220 PRINT#8,N$(ZY)
1230 PRINT#8,Z$(ZY)
1240 PRINT#8,S$(ZY)
1250 PRINT#8,O$(ZY)
1260 PRINT#8:PRINT#8,T$(ZY)
1270 PRINT#8,B$(ZY)
1280 PRINT#8:PRINT#8
1290 ZY=ZY+1:IF N$(ZY)="" THEN GOSUB 5000:GOTO 50
1300 GOTO 1220
1310 'Adressen selektieren
1320 GOSUB 5000
1330 PRINT"*****
*****"
1340 PRINT"***** ADRESSENVERWALTUNG
*** SELEKTIEREN *****"
1350 PRINT"*****
*****"
1360 ZX=0
1370 INPUT#2,"Eingabe des gesuchten Wortes: ";SW$
1380 L=LEN(SW$)
1390 IF RIGHT$(N$(ZX),L)=SW$ OR LEFT$(N$(ZX),L)=SW$
THEN 1470
1400 IF RIGHT$(Z$(ZX),L)=SW$ OR LEFT$(Z$(ZX),L)=SW$
THEN 1470
1410 IF RIGHT$(S$(ZX),L)=SW$ OR LEFT$(S$(ZX),L)=SW$
THEN 1470
1420 IF RIGHT$(O$(ZX),L)=SW$ OR LEFT$(O$(ZX),L)=SW$
THEN 1470
1430 IF RIGHT$(T$(ZX),L)=SW$ OR LEFT$(T$(ZX),L)=SW$
THEN 1470
1440 IF RIGHT$(B$(ZX),L)=SW$ OR LEFT$(B$(ZX),L)=SW$
THEN 1470
1450 ZX=ZX+1:IF ZX<81 THEN 1390
1460 PRINT#2,CHR$(12):INPUT#2,"Dateiende erreicht. W
eiter mit RETURN ";A$:GOSUB 5000:GOTO 50

```

```

1470 GOSUB 5000
1480 PRINT#1,N$(ZX):PRINT#1,Z$(ZX):PRINT#1,S$(ZX):PR
INT#1,O$(ZX)
1490 PRINT#1
1500 PRINT#1,T$(ZX):PRINT#1,B$(ZX):PRINT#2,CHR$(12)
1510 PRINT#2,"Weiterlesen = RETURN  Ändern = A  Dr
ucken = D  Menü = M"
1520 INPUT#2,"Ihre Eingabe bitte :";A$
1530 IF ZX<80 THEN ZX=ZX+1
1540 IF A$="d" OR A$="D" THEN 1570
1550 IF A$="m" OR A$="M" THEN GOSUB 5000:GOTO 50
1560 GOTO 1390
1570 PRINT#8,N$(ZX-1):PRINT#8,Z$(ZX-1):PRINT#8,S$(ZX
-1):PRINT#8,O$(ZX-1)
1580 PRINT#8:PRINT#8,T$(ZX-1):PRINT#8,B$(ZX-1)
1590 PRINT#8:PRINT#8
1600 PRINT#2,CHR$(12):GOTO 1510
1610 'Adressen selektiv drucken
1615 GOSUB 5000
1620 PRINT"*****
*****"
1630 PRINT"***** ADRESSENDATEI
AUSDRUCKEN *****"
1640 PRINT"*****
*****"
1650 PRINT#1,"1  Alphabetisch drucken"
1660 PRINT#1,"2  Einzelne Adressen selektieren"
1670 PRINT#1,"3  Adressen als Anschrift drucken"
1680 PRINT#1,"4  Gesamte Adressendatei drucken"
1690 PRINT#1,"5  Zurück zum Menü"
1700 PRINT#2,"Ihre Eingabe: ";
1710 A$=INKEY$:IF A$="" THEN 1710
1720 IF A$="1" THEN PRINT#2,"Adressen werden alphabe
tisch gedruckt. Bitte Geduld.":GOTO 1780
1730 IF A$="2" THEN 1890
1740 IF A$="3" THEN 2010
1750 IF A$="4" THEN 1220
1760 IF A$="5" THEN GOSUB 5000:GOTO 50
1770 GOTO 1710
1780 'Alphabetisch drucken
1790 FOR AL=65 TO 90
1800 C=1
1810 FOR ZX=0 TO 80
1820 IF LEFT$(N$(ZX),1)=CHR$(AL) OR MID$(N$(ZX),1,1)
=CHR$(AL) THEN GOSUB 1860
1830 C=C+1
1840 NEXT ZX:NEXT AL

```

```

1850 GOSUB 5000:GOTO 50
1860 PRINT#8,N$(ZX):PRINT#8,Z$(ZX):PRINT#8,S$(ZX):PR
INT#8,O$(ZX)
1870 PRINT#8:PRINT#8,T$(ZX):PRINT#8,B$(ZX):PRINT#8:P
RINT#8
1880 RETURN
1890 'Selektieren
1900 ZX=0
1910 INPUT#2,"Eingabe des gesuchten Wortes: ";SW$
1920 L=LEN(SW$)
1930 IF RIGHT$(N$(ZX),L)=SW$ OR LEFT$(N$(ZX),L)=SW$
THEN GOSUB 1860
1940 IF RIGHT$(Z$(ZX),L)=SW$ OR LEFT$(Z$(ZX),L)=SW$
THEN GOSUB 1860
1950 IF RIGHT$(S$(ZX),L)=SW$ OR LEFT$(S$(ZX),L)=SW$
THEN GOSUB 1860
1960 IF RIGHT$(O$(ZX),L)=SW$ OR LEFT$(O$(ZX),L)=SW$
THEN GOSUB 1860
1970 IF RIGHT$(T$(ZX),L)=SW$ OR LEFT$(T$(ZX),L)=SW$
THEN GOSUB 1860
1980 IF RIGHT$(B$(ZX),L)=SW$ OR LEFT$(B$(ZX),L)=SW$
THEN GOSUB 1860
1990 ZX=ZX+1:IF ZX<81 THEN 1930
2000 GOSUB 5000:GOTO 50
2010 'Adressen als Anschrift drucken
2020 ZX=0
2030 FOR N=0 TO 4:PRINT#8:NEXT N
2040 PRINT#8,N$(ZX)
2050 PRINT#8,Z$(ZX)
2060 PRINT#8,S$(ZX)
2070 PRINT#8,O$(ZX)
2080 FOR N=0 TO 5:PRINT#8:NEXT N
2090 REM print#8, datum
2100 REM print#8, Sehr geehrte Frau/sehr geehrter He
rr...
2110 REM print#8, ...Brieftext...
2120 PRINT#8,CHR$(12)
2130 ZX=ZX+1:IF N$(ZX)="" THEN GOSUB 5000:GOTO 50
2140 GOTO 2030
5000 PRINT#0,CHR$(12):PRINT#1,CHR$(12):PRINT#2,CHR$(
12):RETURN
10000 SYMBOL AFTER 91
10010 SYMBOL 91,102,24,60,102,126,102,102,0
10020 SYMBOL 92,195,60,102,102,102,102,60,0
10030 SYMBOL 93,102,0,102,102,102,102,60,0
10040 SYMBOL 123,108,0,120,12,124,204,118,0
10050 SYMBOL 124,0,102,0,60,102,102,60,0

```

10060 SYMBOL 125,0,102,0,102,102,102,62,0
10070 SYMBOL 126,60,102,102,124,102,102,108,224
10080 RETURN

20: Im Unterprogramm 10000 werden einige Tasten für Umlaute undefiniert (vergl. Abb. Seite 15).

30: Die DIMensionierungen für die einzelnen Eingaben (Name, Zusatz usw.).

40: Definition von drei WINDOWS für Kopfzeilen, Eingabe und Menüeinblendungen.

50 bis 120: Das Hauptmenü wird auf den Bildschirm gePRINTet.

130 bis 210: Mit A\$=INKEY\$ wird die Tastatur abgefragt. Je nach Eingabe springt der Rechner in das entsprechende Unterprogramm.

220: Falls Sie eine andere Taste drücken als die zuvor definierten, kehrt der Rechner zur Zeile 130 zurück.

280 bis 430: Hier befindet sich das Unterprogramm zur Eingabe der Adressen. Nach jedem Datenblock haben Sie die Möglichkeit, weiterzuarbeiten oder zum Hauptmenü zurückzukehren (z.B. um die Adressen zu speichern).

440: Wenn 80 Datenblöcke eingegeben wurden, meldet der Rechner "Datei ist voll. Weiter mit RETURN". Sie kehren dann automatisch zum Hauptmenü zurück.

460 bis 600: In diesem Unterprogramm werden die Datenblöcke auf Diskette gespeichert (Kassettenbetrieb ist für diesen Einsatzbereich ungeeignet). Als Hilfsvariable wird ZX gewählt, d.h. die zur Adresseneingabe verwendete Zählvariable Z behält ihren letzten aktuellen Wert. Sie können also nach dem Abspeichern der Adressen mit der Eingabe fortfahren, ohne daß die im Speicher befindlichen Datenblöcke gelöscht werden.

620 bis 720: Hier können Sie eine Adressendatei von Diskette einlesen. Durch die Abfrage, ob der Speicher überschrieben werden soll oder nicht, haben Sie die Möglichkeit, nach der Direkteingabe von Adressen eine externe Datei hinzuzuladen, ohne daß die bisher geschriebenen Datenblöcke gelöscht werden.

Das eröffnet viele bequeme Anwendungen. Sie müssen lediglich darauf achten, daß sich insgesamt nicht mehr als 80 Datenblöcke im Speicher befinden.

730 bis 830: Nach dem Einladen von Adressen erscheint im unteren Bildschirm die Frage, ob Sie die geladene Datei lesen wollen. Wenn Sie hier "j" oder "J" drücken, erscheinen die einzelnen Adressen auf dem Bildschirm. Im anderen Fall gelangen Sie durch Drücken von RETURN (oder irgendeiner anderen Taste) automatisch zum Hauptmenü.

840 bis 910: Wenn Sie bei der Abfrage in Zeile 730 "j" oder "J" gedrückt hatten, erscheint nach dem Auflisten jeder Adresse dieses kleine Auswahlmenü. Hier können Sie weiterlesen, die augenblicklich im Bildschirm sichtbare Adresse ändern, ausdrucken oder zum Hauptmenü zurückkehren.

920: Beim Auflisten der einzelnen Adressen erhöht sich die Hilfsvariable ZX jeweils um Eins. Falls danach ein Adressenstring keine Zeichen enthält (N\$(ZX)=""), so erscheint im Bildschirm der Text "Dateiende erreicht. Weiter mit RETURN". Sie gelangen dann wieder zum Hauptmenü.

940 bis 1100: Dies ist ein Unterprogramm vom Unterprogramm "Adressen lesen". Wenn Sie die Option A (vergl. Zeile 890) gewählt hatten, können Sie hier die aktuelle Adresse ändern bzw. neu eingeben. Auch hier wird die Hilfsvariable ZX genommen und nicht die Hauptzählvariable Z, so daß Sie nach der Änderung normal im Programm fortfahren können.

1310 bis 1450: Mit diesem Adressen-Verwaltungsprogramm haben Sie auch die Möglichkeit, die Datei nach bestimmten Begriffen zu durchforsten. Hierbei ist jedoch eine Einschränkung zu beachten: Es lassen sich nur Wörter selektieren, die am Anfang oder Ende einer Eingabe stehen; z.B. bei dem Namen "Heinz Peter Schenk" entweder "Heinz" oder "Schenk" oder bei der Bemerkung "Mahnung wegen Software" entweder "Mahnung" oder "Software". Im allgemeinen werden Sie damit jedoch gut arbeiten können, da man Adressen in der Regel nach Postleitzahlen, Wohnorten oder Nachnamen selektiert. Das Selektierprinzip ist einfach: Vom eingegebenen Suchwort wird mit $L=LEN(SW)$ die Zeichenlänge ermittelt (Zeile 1380); danach untersuchen die Zeilen 1390 bis 1440, ob sich das Suchwort im rechten oder linken Teil eines Adressenstrings befindet. Im negativen Fall erscheint der Text "Dateiende erreicht. Weiter mit RETURN". Sie gelangen dann ins Hauptmenü zurück.

1470 bis 1600: Falls ein Datenblock das gesuchte Wort enthält, haben Sie hier die Möglichkeit, die Adresse ausdrucken zu lassen, weiterzuselektieren oder ins Hauptmenü zurückzukehren.

1610 bis 2000: Vom Hauptmenü aus können Sie auch ohne den Sprung in ein anderes Unterprogramm gleich zum Drucker Menü übergehen. Bei der Option "Alphabetisch drucken" werden die Nachnamen aller im Speicher befindlichen Adressen untersucht und alphabetisch geordnet ausgedruckt. Seien Sie hier ein wenig geduldig. Ihr CPC-Computer wird mit dem Sortieren in BASIC nur im Dampfmaschinen-Tempo fertig. "Selektiv drucken" ist nützlich, wenn Sie Ihre Adressen z.B. nach Postleitzahlen ausdrucken oder alle überfälligen Rechnungen anmahnen wollen. Je nachdem, welche Eingaben Sie bei "Bemerkung" machen, ist es natürlich auch möglich, alle Geburtstage eines betreffenden Monats auflisten zu lassen. Bei der Option "Gesamtdatetei drucken" werden alle Adressen in der Reihenfolge ausgedruckt, wie sie eingegeben wurden.

2010 bis 2140: Mit der vorliegenden Adressen-Verwaltung können Sie auch Rundschreiben erstellen und den Briefftext mit den Adressen mischen. In dem Unterprogramm ab Zeile 2010 werden zunächst Anschriften in den Briefkopf gedruckt. Danach erfolgen einige Zeilenvorschübe. Hier können Sie nun durch einfache PRINT#-Anweisungen einen Briefftext erstellen und z.B. in der Anrede oder im laufenden Text den Namen des Adressaten einfügen (N\$(ZX)). Zum Abschluß des Briefes erfolgt mit CHR\$(12) ein Formularvorschub. Danach wird die nächste Adresse mit Briefftext gePRINTet usw. Wenn alle Adressen verarbeitet sind, kehrt der Rechner zum Hauptmenü zurück.

Anwender-Entspannung

Dieses Kapitel hat eigentlich nichts mit "Anwenderprogrammen" zu tun; denn es geht hier um Spiele. Aber auch ein ernsthafter Computernutzer kann nicht ständig arbeiten. Hin und wieder tut Zerstreuung gut. Ihr Schneider-Computer ist für Spiele geradezu prädestiniert. Gute Grafik, effektvoller Sound und hohe Rechengeschwindigkeit machen die Computer-Unterhaltung attraktiv.

Sie finden daher nachfolgend ein paar kurzweilige Spielchen, die Sie ruhig einmal "zwichenschieben" sollten, wenn Ihnen danach zumute ist. Wie Sie sehen, lassen sich auch mit wenigen Programmzeilen passable Ergebnisse erzielen.

Das "Lottospiel" können Sie gleich zweifach nutzen: 1) simuliert es das Ziehen von Lottozahlen und vergleicht sie mit dem eingegebenen Tip und 2) können Sie das Programm zum Ermitteln von Lottozahlen einsetzen, die Sie dann in einen echten Tip-schein eintragen (wenn Sie nicht gewinnen, war der Computer schuld). Auch als Partyspaß ist das "Lottospiel" geeignet.


Beim "Nervenkitzel" geht es um Leben und Tod. Hier werden Sie aufgefordert, zwischen sechs Pistolen zu wählen, von denen eine geladen ist. Wenn Sie Pech haben meldet der Rechner "(Name) ist heute bedauerlicherweise gestorben". Aber zum Glück werden Sie nach Beendigung des Programms wiedergeboren. "Nervenkitzel" ist also eine Variante vom "Russischen Roulette".

"Supermind" ist etwas für fleißige Denker. Die Namensverwandtschaft mit "Mastermind" ist rein beabsichtigt. Gegenüber dem Plastikbrettchen mit den vielen Steckern hat das Computerprogramm den Vorteil, daß Sie eine beliebige Anzahl von Steckern oder Farben frei wählen können (beginnen Sie aber nicht gleich mit dem höchsten Schwierigkeitsgrad).

Als reines Glücksspiel finden Sie anschließend noch "Amerikanisches Würfeln", eine Computer-Simulation von "Crabs". Hier können Sie allein gegen den Rechner oder mit insgesamt zehn Personen spielen.



Lottospiel

Der Computer ermittelt aus 49 "Kugeln" sechs Zufallsergebnisse plus Zusatzzahl. Durch eine Vergleichsroutine kann jede Zahl nur einmal vorkommen.




Wenn Sie das Spiel nur zum Ermitteln von Lottozahlen für einen echten Tipschein einsetzen wollen, drücken Sie hier jeweils RETURN.

Im Gegensatz zum echten Lottospiel ist bei dieser Computer-version auch die Spielgebühr ein Lotteriespiel für sich. Der Computer ermittelt per Zufalle eine mehr oder weniger hohe Geldsumme.



Nachdem die Zahlen gezogen sind, vergleicht der Rechner das Ergebnis mit Ihrem Tip. Wenn Sie mehr als drei Richtige haben, wird ein entsprechender Gewinn ausgeschüttet, den Sie auf Ihr Schweizer Nummernkonto überweisen können. Von diesem Konto werden auch die Spielgebühren abkassiert.



```

10 'Lottospiel
20 DIM ST(6),CT(6)
30 KO=0
40 MODE 1
50 WINDOW#0,1,40,1,4:WINDOW#1,1,40,5,25
60 PRINT"*****"
70 PRINT"***** L O T T O S P I E L *****"
80 PRINT"*****"
90 RANDOMIZE KO
100 PRINT#1,CHR$(12)
110 PRINT#1,"Geben Sie Ihren Tip ein:"
120 PRINT#1
130 FOR X=1 TO 6
140 PRINT#1,X;". = ";
150 INPUT#1,ST(X-1)
160 NEXT X
170 PRINT#1
180 PRINT#1:PRINT#1,"Die Spielgebuehr betraegt in di
eser"
190 GB=INT(10*RND(1))+1
200 KO=KO-GB
210 PRINT#1,"Runde";GB;"DM. Viel Glueck.":PRINT#1
220 'Kugeln ziehen
230 FOR X=0 TO 6
  
```

```

240 KU=INT(49*RND(1))+1
250 FOR PR=0 TO KU
260 IF CT(X)=KU THEN 240
270 NEXT PR
280 SOUND 1,3000,150,7,0,0,1
290 IF X<6 THEN PRINT#1,X;". Kugel = ";KU
300 IF X>5 THEN PRINT#1:PRINT#1,"Zusatzzahl = ";KU
310 CT(X)=KU
320 FOR Z=0 TO 1500:NEXT Z: 'Verzoegerungs-Schleife
330 NEXT X
340 'Ermittlung von richtigen Tips
350 Z=0
360 FOR X=0 TO 6
370 FOR Y=0 TO 5
380 IF ST(X)=CT(Y) THEN Z=Z+1
390 NEXT Y
400 NEXT X
410 PRINT#1:PRINT#1,"Sie haben";Z;"Richtige."
420 IF Z<3 THEN PRINT#1:PRINT#1,"Tut mir leid.":GOSU
B 530
430 IF Z>2 THEN GOSUB 460
440 IF A$="J" OR A$="j" THEN 90
450 MODE 1:END
460 'Drei oder mehr Richtige
470 PRINT#1:PRINT#1,"Herzlichen Glueckwunsch. Sie ha
ben"
480 DM=3*INT((Z^Z)*100)/100
490 PRINT#1,DM;"DM gewonnen."
500 PRINT#1:PRINT#1,"Soll das Geld auf Ihr Schweizer
"
510 INPUT#1,"Nummernkonto ueberwiesen werden (J/N)";
A$
520 IF A$="J" OR A$="j" THEN KO=KO+DM
530 PRINT#1:PRINT#1,"Ihr Kontostand liegt bei ";KO;"
DM"
540 PRINT#1:INPUT#1,"Noch ein Spiel ";A$
550 RETURN

```

20: Für den Spieler-Tip (Variable ST(x)) und den Computer-Tip (Variable CT(x)) werden DIMensionierungen vorgenommen.

90: Dieser Befehl sorgt dafür, daß bei mehreren Programmdurchgängen immer eine andere Folge von Zufallszahlen erzeugt wird. Der Computer benutzt nämlich bei einem RND-Befehl gar keine

echten Zufallszahlen, sondern eine festgelegte Reihenfolge. Sie können das einmal ausprobieren, indem Sie Zeile 90 fortlassen und mehrere Lottoziehungen durchführen lassen. Sie werden sehen, daß die Zahlen immer identisch sind.

100 bis 160: Hier können Sie Ihren Tip eingeben. Falls Sie das Programm gerne mit mehreren Personen spielen wollen, sollten Sie eine zusätzliche Abfrage einbauen; z.B. in Zeile

```
95 INPUT "Wie viele Personen nehmen an dem Spiel teil";P
96 FOR Y=1 TO P
165 NEXT Y
```

In diesem Fall wäre es sinnvoll, aus ST(6) in Zeile 20 eine zweidimensionale Feldvariable zu machen; z.B. ST(11,6), so daß bis zu zwölf Personen mitspielen können. Zeile 150 müßte dann ebenfalls geändert werden durch den Zusatz ST(Y,X-1).

180 bis 210: Die Gebühr für eine Spielrunde kann bis zu DM 10,- kosten. Der Betrag wird durch $KO=KO-GB$ vom Konto abgezogen. Der Kontostand ist zum Spielbeginn=0. Wenn Sie also in der ersten Runde nicht gewinnen, haben Sie ein Minus auf dem Konto.

220 bis 330: Hier findet die eigentliche "Ziehung" statt. Da bei dem normalen RND-Befehl nur Zahlen zwischen 0 und 1 erzeugt werden, muß die Anweisung lauten $KU=INT(49*RND(1))+1$. Die FOR...NEXT-Schleife ab Zeile 250 ist eine Prüfroutine, in der alle neu ermittelten Zahlen mit den bisherigen verglichen werden. Wenn eine Übereinstimmung vorliegt, wird der "Ziehungsvorgang" einfach wiederholt. (Zum Spaß können Sie diese Prüfroutine einmal fortlassen und möglicherweise zweimal die gleiche Zahl als Ergebnis auf dem Bildschirm sehen.) Zeile 280 spielt einen kurzen Ton, sobald eine Zahl gezogen wurde. Die FOR...NEXT-Schleife in Zeile 320 dient nur zur Verzögerung. Es ist weniger spannend, wenn alle Lottozahlen innerhalb einer halben Sekunde auf dem Bildschirm präsentiert werden.



340 bis 400: Nach der Ziehung müssen Sie nicht mühsam vergleichen, ob Sie drei oder mehr "Richtige" getippt hatten. Diese Arbeit besort der Rechner. Zeile 380 stellt die Bedingung "Wenn der Spieler-Tip mit dem Computer-Tip übereinstimmt, dann zähle der Variable Z Eins hinzu".


410: Hier wird ausgegeben, wieviele Zahlen zwischen Spieler-Tip und Computer-Tip übereinstimmen (Z).

460 bis 520: Wenn Sie drei oder mehr "Richtige" haben, springt der Rechner in dieses Unterprogramm. Hier erfahren Sie die Höhe Ihres Gewinns (die Berechnungsformel in Zeile 480 können Sie beliebig ändern). Anschließend wird der Betrag auf Ihr Konto gebucht und der aktuelle Kontostand angezeigt.

Nervenkitzel

Bei diesem Spiel geht es um Leben und Tod. Falls Sie ängstlich sind, geben Sie zum Spielbeginn nicht Ihren eigenen Namen ein, sondern den Ihres "ärgersten Feindes" ein. Das beruhigt unheimlich.





```
10 'Nervenkitzel
20 GOSUB 390
30 MODE 1
40 WINDOW#0,1,40,1,4:WINDOW#1,1,40,5,25
50 N=RND(1)
60 PRINT"*****"
70 PRINT"***** N E R V E N K I T Z E L *****"
80 PRINT"*****"
90 PRINT#1,"Geben Sie Ihren Namen ein:"
100 INPUT#1,N$
110 PRINT#1,CHR$(12)
120 PRINT#1,"Guten Tag ";N$:PRINT#1
130 PRINT#1,"Sie haben die Wahl zwischen sechs"
140 PRINT#1,"Pistolen. Eine davon ist geladen."
150 PRINT#1
160 INPUT#1,"Welche Pistole wählen Sie ";P
170 RANDOMIZE N
180 S=INT(5*RND(1))+1
190 IF S=P THEN 320
200 PRINT#1,CHR$(7)
210 PRINT#1,CHR$(12)
220 GL=GL+1
230 PRINT#1,"Sie leben noch."
240 PRINT#1
```

```

250 PRINT#1,"Sie haben nun";GL;"mal Glück gehabt."
260 PRINT#1
270 PRINT#1,"Wollen Sie es noch einmal"
280 PRINT#1,"riskieren (J/N)?"
290 A$=INKEY$:IF A$="" THEN 290
300 IF A$="J" OR A$="j" THEN PRINT#1,CHR$(12):N=N+S:
GOTO 130
310 MODE 1:END
320 FOR X=1000 TO 1500 STEP 50
330 SOUND 1,X,10,7
340 NEXT X
350 MODE 1
360 PRINT N$;" ist heute"
370 PRINT"bedauerlicherweise verstorben."
380 END
390 SYMBOL AFTER 91
400 SYMBOL 91,102,24,60,102,126,102,102,0
410 SYMBOL 92,195,60,102,102,102,102,60,0
420 SYMBOL 93,102,0,102,102,102,102,60,0
430 SYMBOL 123,108,0,120,12,124,204,118,0
440 SYMBOL 124,0,102,0,60,102,102,60,0
450 SYMBOL 125,0,102,0,102,102,102,62,0
460 SYMBOL 126,60,102,102,124,102,102,108,224
470 RETURN

```

120 bis 160: Der erste Bildschirm wird geprintet.

170: Auch hier soll, genau wie im "Lottospiel" (vergl. Zeile 90) bei jedem Programmdurchlauf eine andere Reihenfolge von Zufallszahlen erzeugt werden. Das "Todesspiel" wäre sonst kalkulierbar.


180 und 190: S ist die Variable für den Schuß-Revolver. Wenn die ermittelte Zahl mit der von Ihnen angegebenen übereinstimmt (IF S=P), dann springt der Rechner zur Zeile 320.

200: Stimmt die Computer-Zahl nicht mit Ihrer Eingabe überein, wird das Programm hier fortgesetzt. Anstelle eines fauchenden Schußgeräuschs erklingt nur ein Beep. Dafür erscheint die gute Nachricht, daß Sie noch leben. Sie können, falls Sie daran interessiert sind, das Programm noch erweitern, indem Sie die Nummer der "geladenen" Pistole ausdrucken lassen (Variable S).

360 und 370: Hier ereilt Sie die Nachricht, daß Sie soeben verstorben sind.

Supermind

Gegenüber der Brettspiel-Version von Mastermind wählt der Computer keine echten Farben, sondern Ziffern. Beim Spielen macht das keinen Unterschied.



Geben Sie Ihren Tip ohne Zwischenräume oder Kommas ein (z.B. 223 oder 132). Auf dem Bildschirm wird dann angezeigt, ob Sie einen Treffer hatten (T), eine Farbe oder Position richtig wählten (H) oder ob gar nichts richtig war (N). Sie haben unbegrenzt viele Versuche.

```
10 'Supermind  
20 GOSUB 470  
30 MODE 1  
40 WINDOW#0,1,40,1,4:WINDOW#1,1,40,5,25  
50 PRINT"*****"  
60 PRINT"***** S U P E R M I N D *****"  
70 PRINT"*****"
```

```

80 N=RND(1):RANDOMIZE N
90 INPUT#1,"Wählen Sie die Anzahl der Stecker";S
100 PRINT#1
110 INPUT#1,"Wählen Sie die Anzahl der Farben";F
120 N=RND(1)
130 FOR X=1 TO S
140 Z=INT(F*RND(1))+1
150 FOR PR=1 TO Z
160 IF N(PR)=Z THEN 140
170 NEXT PR
180 N(X)=Z
190 NEXT X
200 PRINT#1,CHR$(12)
210 PRINT#1,"Eingabe für";S;"Stecker mit den"
220 PRINT#1,"Farben 1 bis";F
230 PRINT#1
240 PRINT#1,"          T";"  "; "H";"  "; "N"
250 INPUT#1,W$
260 A=0:B=0
270 FOR X=1 TO S
280 C(X)=VAL(MID$(W$,X,1))
290 FOR PR=1 TO S
300 IF N(PR)<>C(X) THEN 340
310 IF X<>PR THEN 330
320 A=A+1
330 B=B+1
340 NEXT PR
350 NEXT X
360 PRINT#1,"          ";B;A-T;A
370 PRINT#1
380 IF A<S THEN 250
390 PRINT#1
400 PRINT#1,"Stimme alles. Sie sind wirklich ein"
410 PRINT#1,"Superhirn."
420 PRINT#1:PRINT#1
430 PRINT#1,"Noch ein Spiel (J/N)?"
440 A$=INKEY$:IF A$="" THEN 440
450 IF A$="J" OR A$="j" THEN PRINT#1,CHR$(12):GOTO 8
0
460 MODE 1:END
470 SYMBOL AFTER 91
480 SYMBOL 91,102,24,60,102,126,102,102,0
490 SYMBOL 92,195,60,102,102,102,102,60,0
500 SYMBOL 93,102,0,102,102,102,102,60,0
510 SYMBOL 123,108,0,120,12,124,204,118,0
520 SYMBOL 124,0,102,0,60,102,102,60,0
530 SYMBOL 125,0,102,0,102,102,102,62,0

```

540 SYMBOL 126,60,102,102,124,102,102,108,224
550 RETURN

130 bis 190: Hier werden die Farben und Positionen der Stecker per Zufall festgelegt. Eine Prüfroutine (150 bis 170) sorgt dafür, daß kein logischer Fehler auftritt und jede Farbe/Stecker-Kombination nur einmal vorkommt.

270 bis 350: Hier wird verglichen, inwieweit Ihre Eingaben mit den Werten des Rechners übereinstimmen. Dabei wird W\$ durch die MID\$-Funktion Zeichen für Zeichen gelesen und durch VAL in eine numerische Variable umgewandelt (im anderen Fall müßten Sie drei oder mehr Eingaben machen und jedesmal RETURN drücken).

380: Solange Ihre Kombinationen nicht mit den Computer-Werten übereinstimmen, kehrt das Programm zur Zeile 250 zurück, wo Sie einen weiteren Versuch haben.

Amerikanisches Würfeln


Das folgende Spiel kennen Sie sicher aus amerikanischen Spielfilmen: Abenteuerliche Gestalten stehen dicht gedrängt um einen Spieltisch. Alle Augen sind auf den Held gerichtet, der (mit einer Blondine, die ihm Glück bringen soll) im entscheidenden Wurf alles riskiert (und natürlich gewinnt). Die Leute klatschen Beifall, reißen die Hüte hoch, und der Held sagt lässig "Komm Baby, wir geh'n ´was trinken"...

Das Spiel heißt "Crabs" und läßt sich auch auf dem Computer gut simulieren. In der vorliegenden Programmversion können Sie allein gegen den Computer oder mit insgesamt zehn Personen spielen. Die Regeln sind schnell erklärt:


Wer im ersten Wurf 7 oder 11 würfelt, gewinnt.

Wer 2,3 oder 12 würfelt, verliert.


Wer im ersten Wurf keine der beiden Bedingungen erfüllt, würfelt so lange weiter, bis er entweder seinen ersten Wurf ("Punkt") wiederholt (= gewonnen) oder eine 7 würfelt (= verloren).



Zum Würfeln drücken Sie irgendeine Taste. Es ertönt dann ein würfelartiges Geräusch. Die ermittelte Zahl wird auf dem Bildschirm angezeigt. Je nach Ergebnis würfelt der betreffende Spieler weiter oder scheidet mit Gewinn bzw. Verlust aus.



Nach jeder Runde können Sie sich durch Drücken von RETURN den Spielstand einblenden lassen und ggf. weiterspielen.



```

10 'Amerikanisches Würfeln
20 GOSUB 630
30 MODE 1
40 WINDOW#0,1,40,1,4:WINDOW#1,1,40,5,25
50 PRINT"*****"
60 PRINT"***** AMERIKANISCHES WÜRFELN *****"
70 PRINT"*****"
80 INPUT#1,"Anzahl der Spieler: ";AN
90 PRINT#1:PRINT#1
100 INPUT#1,"Weiter mit RETURN";A$
110 PRINT#1,CHR$(12)
120 PRINT#1,"Zum Würfeln irgendeine Taste drücken"
130 FOR X=1 TO AN
140 PRINT#1
150 PRINT#1,"Spieler";X;"würfelt jetzt"
160 A$=INKEY$:IF A$="" THEN 160
170 GOSUB 600
180 RANDOMIZE W1+X
190 W1=INT(6*RND(1))+1
200 RANDOMIZE W2+W1+X
210 W2=INT(6*RND(1))+1
220 WW=W1+W2
230 PRINT#1,"          Würfelzahl: ";WW;
240 IF WW=7 OR WW=11 THEN 390
250 IF WW=2 OR WW=3 THEN 420
260 PRINT#1," Punkt"
270 B=WW
280 PRINT#1:PRINT#1,"Spieler";X;"würfelt"
290 A$=INKEY$:IF A$="" THEN 290
300 GOSUB 600
310 RANDOMIZE W1+X
320 W1=INT(6*RND(1))+1
330 RANDOMIZE W2+W1+X
340 W2=INT(6*RND(1))+1
350 WW=W1+W2
360 PRINT#1,"          Würfelzahl: ";WW;
370 IF WW<>7 AND WW<>B THEN 280
380 IF WW=7 THEN 420
390 S(X)=S(X)+1
400 PRINT#1,"Gewonnen"
410 GOTO 440
420 S(X)=S(X)-1
430 PRINT#1,"Verloren"
440 NEXT X
450 PRINT#1:PRINT#1
460 INPUT#1,"Spielstand mit RETURN";A$
470 PRINT#0,CHR$(12):PRINT#1,CHR$(12)

```

```

480 PRINT"*****"
490 PRINT"***** S P I E L S T A N D *****"
500 PRINT"*****"
510 PRINT#1,"Spieler","Punkte"
520 FOR X=1 TO AN
530 PRINT#1,X,S(X)
540 NEXT X
550 PRINT#1:PRINT#1
560 INPUT#1,"Noch eine Runde ";A$
570 IF A$="J" OR A$="j" THEN PRINT#0,CHR$(12):PRINT#
1,CHR$(12):GOTO 50
580 MODE 1
590 END
600 'Würfelsound
610 SOUND 1,3000,7,0,0,1
620 RETURN
630 SYMBOL AFTER 91
640 SYMBOL 91,102,24,60,102,126,102,102,0
650 SYMBOL 92,195,60,102,102,102,102,60,0
660 SYMBOL 93,102,0,102,102,102,102,60,0
670 SYMBOL 123,108,0,120,12,124,204,118,0
680 SYMBOL 124,0,102,0,60,102,102,60,0
690 SYMBOL 125,0,102,0,102,102,102,62,0
700 SYMBOL 126,60,102,102,124,102,102,108,224
710 RETURN

```

130 bis 440: Die Anzahl der FOR...NEXT-Durchläufe (AN) richtet sich nach der Anzahl der Spieler (AN). Genau wie im "Lotto-spiel", wo immer wieder eine neue Folge von Zufallszahlen ermittelt werden muß, wird auch hier der Befehl RANDOMIZE=n eingesetzt (Zeile 180 und 200). Die Ergebnisse von Würfel 1 und Würfel 2 werden addiert (WW=W1+W2) und auf den Bildschirm gePRINTet (Zeile 230). Danach vergleicht der Rechner, ob WW 7,11,2,3 oder 12 ist. In den ersten beiden Fällen hätte der Spieler gewonnen, ansonsten verloren. Trifft keine der Bedingungen aus den Zeilen 240 und 250 zu, wird die Würfelzahl mit dem Zusatz "Punkt" versehen (Zeile 260). Danach wiederholt sich der Würfelvorgang so lange, bis entweder das Ergebnis von "Punkt" zutrifft oder eine 7 gewürfelt wird. In der Regel ist das nach vier bis fünf Würfeln positiv oder negativ geschafft.



470 bis 550: Nach jeder Runde können Sie durch Drücken von RETURN den aktuellen Spielstand auflisten lassen. Durch eine

FOR...NEXT-Schleife erscheint die Nummer des Spielers und die Zahl der errungenen Siege (oder Verluste). Wenn Sie neu beginnen und gleich verlieren, haben Sie -1.

560 bis 590: Nach dem Auflisten des Punktestandes können Sie weitere Runden spielen. Die vorherigen Punkte werden dabei nicht gelöscht.

Musik-Box

Falls Sie sich gelegentlich mit Musik entspannen möchten, können Sie auch hierfür Ihren Schneider-Computer heranziehen. Wie Sie sicher wissen, lassen sich beim CPC bis zu drei Melodiestimmen gleichzeitig programmieren.



Es ist jedoch etwas mühsam, für ein komplettes Musikstück alle Notenwerte einzeln in SOUND-Anweisungen einzugeben. Mit dem folgenden Programm können Sie diese Aufgabe wesentlich bequemer lösen. Die Musik-Box erlaubt das fortlaufende Eingeben von Notenwerten, die jederzeit gespielt und verändert werden können. Ein fertiggestelltes Musikstück kann abgespeichert und eingeladen werden, wann immer Sie es wünschen.

```

10 'Musik-Box
20 MODE 2
30 WINDOW#0,1,80,1,20:WINDOW#1,1,80,21,25
40 DIM K(100),F(100),D(100),L(100)
50 X=0:VK=1:VL=5
60 PRINT"*****
*****"
70 PRINT"*****  M U S I K - B
   O X *****"
80 PRINT"*****
*****"
90 PRINT:PRINT
100 PRINT"          1.  Notenwerte eingeben"
110 PRINT"          2.  Notenwerte spielen"
120 PRINT"          3.  Notenwerte abspeichern"
130 PRINT"          4.  Notenwerte einladen"
140 PRINT"          5.  Editieren"
150 PRINT
160 PRINT"          6.  Programm beenden"
170 INPUT#1,"Ihre Eingabe: ",E
180 ON E GOTO 200,360,430,530,660,190
190 MODE 2:END
200 'Notenwerte eingeben
210 PRINT#0,CHR$(12):PRINT#1,CHR$(12)
220 PRINT#1,"Mit Kanalangabe <5> Noteneingabe beenden
"
230 CLS
240 X=X+1:PRINT:PRINT"Nr.";X
250 PRINT"Eingabe Kanal-Nr. (1 bis 3) oder RETURN, we
nn Kanal-Nr.=";VK;:INPUT K(X)
260 IF K(X)=4 THEN PRINT#0,CHR$(12):PRINT#1,CHR$(12):
GOTO 60
270 IF K(X)=5 THEN K(X+1)=5:PRINT#0,CHR$(12):PRINT#1,
CHR$(12):X=X-1:GOTO 60
280 IF K(X)=0 THEN K(X)=VK
290 PRINT"Eingabe Frequenzwert oder RETURN, wenn Freq
uenz=";F(X-1);:INPUT F(X)

```

```

300 IF F(X)=0 THEN F(X)=F(X-1)
310 PRINT"Eingabe Tondauer oder RETURN, wenn Tondauer
=";D(X-1);:INPUT D(X)
320 IF D(X)=0 THEN D(X)=D(X-1)
330 PRINT"Eingabe Lautstärke oder RETURN, wenn Lautst
ärke=";VL;:INPUT L(X)
340 IF L(X)=0 THEN L(X)=VL
350 GOTO 240
360 'Notenwerte spielen
370 Z=1
380 PRINT#1,CHR$(12):PRINT#1,"Notenwerte werden gespi
elt"
390 SOUND K(Z),F(Z),D(Z),L(Z)
400 Z=Z+1
410 IF K(Z)=5 THEN CLS:PRINT#1,CHR$(12):GOTO 60
420 GOTO 390
430 'Notenwerte abspeichern
440 PRINT#1,CHR$(12)
450 INPUT#1,"Eingabe Dateiname: ";DN$
460 PRINT#1,"Daten werden gespeichert. Bitte warten."
470 Z=1
480 OPENOUT DN$
490 PRINT#9,K(Z):PRINT#9,F(Z):PRINT#9,D(Z):PRINT#9,L(
Z)
500 Z=Z+1
510 IF K(Z)=5 THEN CLOSEOUT:PRINT#1,CHR$(12):GOTO 170
520 GOTO 490
530 'Notenwerte einladen
540 PRINT#1,CHR$(12)
550 INPUT#1,"Eingabe Dateiname: ";DN$
560 PRINT#1,"Daten werden eingelesen. Bitte warten."
570 Z=1
580 OPENIN DN$
590 IF EOF THEN 630
600 INPUT#9,K(Z),F(Z),D(Z),L(Z)
610 Z=Z+1
620 GOTO 590
630 K(Z+1)=5
640 CLOSEIN
650 PRINT#1,CHR$(12):GOTO 170
660 'Editieren
670 CLS:PRINT#1,CHR$(12)
680 FOR Y=1 TO 19:PRINT Y;"=";K(Y);F(Y);D(Y);L(Y):NEX
T Y
690 FOR Y=20 TO 38:LOCATE 20,Y-19:PRINT Y;"=";K(Y);F(
Y);D(Y);L(Y):NEXT Y
700 FOR Y=39 TO 57:LOCATE 40,Y-38:PRINT Y;"=";K(Y);F(

```

```

Y);D(Y);L(Y):NEXT Y
710 FOR Y=58 TO 76:LOCATE 60,Y-57:PRINT Y;"=";K(Y);F(
Y);D(Y);L(Y):NEXT Y
720 INPUT#1,"Weiter mit RETURN. Editieren mit <E>";E
$
730 IF E$="E" OR E$="e" THEN 740 ELSE 840
740 PRINT#1,CHR$(12)
750 INPUT#1,"Eingabe-Nr.";Z
760 INPUT#1,"Neue Kanal-Nr. ";K(Z)
770 INPUT#1,"Neue Frequenz ";F(Z)
780 INPUT#1,"Neue Tondauer ";D(Z)
790 INPUT#1,"Neue Lautstärke";L(Z)
800 PRINT#1,CHR$(12)
810 INPUT#1,"Weitereditieren mit RETURN. Zum Menü mit
<M> ";E$
820 IF E$="M" OR E$="m" THEN CLS:PRINT#1,CHR$(12):GOT
O 60
830 GOTO 740
840 FOR Y=77 TO 95:LOCATE 1,Y-76:PRINT Y;"=";K(Y);F(Y
);D(Y);L(Y):NEXT Y
850 FOR Y=96 TO 100:LOCATE 20,Y-95:PRINT Y;"=";K(Y);F
(Y);D(Y);L(Y):NEXT Y
860 PRINT#1,CHR$(12)
870 GOTO 720

```

40: Die vorliegende Programmversion erlaubt die Eingabe von 100 Notenwerten. Diese Zahl kann wahlweise erhöht werden.

50: Zum Programmbeginn wird die Kanalnummer mit 1 und die Lautstärke mit 5 vorgegeben. Wenn Sie diese Werte übernehmen möchten, brauchen Sie bei der späteren Abfrage jeweils nur RETURN zu drücken.

60 bis 190: Das Hauptmenü. Wenn Sie aus einem Unterprogramm ins Hauptmenü zurückkehren, können Sie z.B. die Noteneingabe fortsetzen, ohne daß die bisherigen Werte gelöscht werden. Der Rechner speichert die jeweils letzte Eingabenummer (Variable X).

200 bis 350: Mit diesem Unterprogramm erfolgt die Noteneingabe. Es können die Kanalnummer, Frequenz, Tondauer und Lautstärke bestimmt werden. Die Frequenzwerte zu den einzelnen Noten entnehmen Sie der Bedienungsanleitung im Anhang VII, den Seiten 1 bis 3. Ein besonderer Komfort dieses Programms besteht darin, daß der Rechner die jeweils vorhergehenden Werte

für die nächste Eingabe als Option bereithält. Sie können entweder durch Drücken von RETURN diese Werte in die neue Noteneingabe übernehmen oder andere Eingaben machen. Wenn Sie z.B. bei der Eingabe Nr. 1 die Tondauer mit 100 angegeben haben, fragt der Rechner bei Nr. 2 "Eingabe Tondauer oder RETURN, wenn Tondauer = 100". Auf diese Weise läßt sich die Noteneingabe sehr angenehm gestalten, wenn viele Noten mit ähnlichen Werten programmiert werden.

360 bis 420: Hier wird das Musikstück gespielt. Als Zählvariable dient Z, so daß die Noteneingabe nach dem Spiel mit dem aktuellen Wert von X fortgesetzt werden kann. Soll mehrstimmiges Spiel erfolgen, so müssen auch die zeitweiligen Pausen der Tonkanäle miteingegeben werden. Unterbleibt dies, kommt es zu Klangverschiebungen, da der Rechner die Eingaben kontinuierlich abarbeitet und nicht wartet, bis ein Tonkanal zu Ende gespielt hat.

430 bis 520: Nach der Eingabe eines Dateinamens werden alle bislang vorgegebenen Notenwerte auf Kassette oder Diskette abgespeichert.

530 bis 650: Vom Hauptmenü aus kann ein abgespeichertes Musikstück in den Rechner geladen werden. Danach läßt es sich im Editiert-Modus bearbeiten oder einfach spielen.

660 bis 870: Die einzelnen Notenwerte eines Musikstücks lassen sich im Editier-Modus beliebig verändern. Zunächst erscheinen hier alle Eingabenummern (1 bis 100) und die dazugehörenden Notenwerte. Danach können Sie einzelne Nummern aufrufen und die Frequenzangabe, Tondauer usw. erneuern. Durch Drücken von M oder m gelangen Sie wieder zum Hauptmenü.

Peripheres

Bevor Sie irgendwo eine Kontaktanzeige aufgeben, um für Ihren Schneider-Computer passenden Anschluß zu finden, sollten Sie erst einmal dieses Kapitel lesen.

Beim praktischen Arbeiten mit dem CPC 464/664 ist man nicht ausschließlich auf Peripheriegeräte vom Hersteller angewiesen. Der 8-Bit Parallelausgang (Centronics) erlaubt den Anschluß zahlreicher, leistungsfähiger Peripheriegeräte. Wir haben für Sie einige dieser Geräte herausgesucht und getestet; denn im Laufe der Zeit werden Sie sicher den Wunsch haben, Ihr System zu erweitern. Ein Joystick z.B. erlaubt die bequeme Steuerung von Bildschirmfiguren (in Anwenderprogrammen und Spielen). Schließlich werden Sie auch auf einen Drucker nicht verzichten wollen. Zum Ausdrucken von Listings genügt meist ein preiswerter Thermodrucker. Höhere Druckgeschwindigkeit, verschiedene Ausdruckarten (fett, gedehnt usw.) und Grafikfähigkeit bietet ein Matrix-Nadeldrucker. Mit ihm lassen sich große Text- und Datenmengen verarbeiten. Wem es um ein einwandfreies Schriftbild und eine große Auswahl an unterschiedlichsten Schrifttypen ankommt, der wird auf einen Typenraddrucker zugreifen.

An den Schneider-Computer läßt sich natürlich auch ein Plotter anschließen. Für professionelle Anwendungen im grafischen Bereich ist dies sinnvoll. Auch in diesem Buch konnten Sie ein Programm für einen praktischen Plottereinsatz finden (Seite 71, "Angebots-Plotter").

Die auf den folgenden Seiten vorgestellten Modelle bilden eine unvollständige Auswahl von Peripheriegeräten, die sich ohne weitere Zusatzteile (wie Spezialkabel) an den CPC 464/664 anschließen lassen. Es sind jedoch keineswegs die einzigen, die im Betrieb mit dem Schneider-Computer problemlos funktionieren. Insofern stellen die Rezensionen keine ausschließliche Qualitätshervorhebung dar.



Competition-Pro

Joystick

Der Competition-Pro ist den Spielhallen-Joysticks nachempfunden und liegt sehr gut in der Hand. Er besitzt GummifüÙe, die für festen Stand sorgen. Praktisch und benutzerfreundlich ist das extra lange Anschlußkabel. Durch die Anbringung von zwei Feuerknöpfen am Fuß des Joysticks wurde an Rechts- und Linkshänder gedacht.

Da der CPC 464/664 nur ein Joystickport besitzt (ein zweites Port befindet sich am Fuß des Original Schneider-Joysticks), kann der Competition-Pro nur eingesetzt werden, wenn man auf einen zweiten Joystick verzichtet. In diesem Fall ist der Competition-Pro wegen seines günstigen Preises eine empfehlenswerte Alternative zum teuren Schneider-Joystick.

Informationen: Dynamics, Postfach 11 20 05
2000 Hamburg 11



Centronics CLP

Matrix-Nadeldrucker

Wenn Sie einen Drucker nicht für umfangreiche Textmengen oder Korrespondenz einsetzen, sondern nur Ihre Listings und private Texte ausdrucken wollen, genügt ein kleiner Matrix-Nadeldrucker.

Der Centronics GLP ist nahezu baugleich mit dem NLQ-401 von Schneider und arbeitet mit 50 Zeichen pro Sekunde. Im Schönschriftmodus bringt es es auf 12 Zeichen pro Sekunde. Mit dem GLP lassen sich 80 Zeichen pro Zeile in Pica-Schrift (entspr. 10 Zeichen/Zoll) ausdrucken. Daneben gibt es noch drei andere Schriftarten.

Bit-Image-Grafik (Einzelpunkt-Grafik) ist mit dem GLP ebenfalls möglich. Die Punktdichte kann dabei 480, 960 und 1920 Punkte pro 8-Zoll-Breite betragen. Dadurch können mit dem GLP auch Hardcops vom Bildschirm erstellt werden.

Informationen: Jürgen Schumpich GmbH, Postfach 6352
8012 Ottobrunn



STAR radix-10

Matrix-Nadeldrucker

Der radix-10 ist ein robuster und bedienungsfreundlicher Matrix-Nadeldrucker, der für seine Preisklasse viel Leistung bietet: 200 Zeichen pro Sekunde (im NLQ-Schönschriftmodus 32 Zeichen pro Sekunde), 16 kByte Pufferspeicher, 6 internationale Zeichensätze, 2 mal 96 frei definierbare Zeichen (Download-Characters), verschiedene Schriftarten (10, 12 und 17 Zeichen pro Zoll), Proportionalschrift, Halbfett-, Fett- und Microschrift, bidirektionaler und unidirektionaler Ausdruck (für extrem sauberes Druckbild).

Außerdem ermöglicht der radix-10 Bit-Image-Grafik mit 480, 960 und 1920 Punkten pro Zeile. Das Gerät verarbeitet Endlosformulare und Einzelblätter, wobei Einzelblätter automatisch eingezogen werden.

Informationen: STAR Europé GmbH, Frankfurter Allee 1-3
6236 Eschborn




Qume LetterPro-20

Typenraddrucker

Wenn Sie Ihren CPC-Computer professionell für Textverarbeitung oder berufliche Korrespondenz einsetzen wollen, wirkt ein Matrix-Text unpersönlich und ist z.T. auch schlecht lesbar. Außerdem verfügen Matrix-Nadeldrucker nur über eine begrenzte Zahl von verschiedenen Schriftarten.

Beim Typenraddrucker LetterPro-20 stehen Ihnen insgesamt über 120 Schrifttypen zur Verfügung. Das Gerät arbeitet mit 20 Zeichen pro Sekunde und ist durch die Ansteuerung von 120 Punkten horizontal und 48 Punkten vertikal im gewissen Rahmen grafikfähig. Wahlweise lassen sich Einzelblätter oder Endlosformulare (mit aufsetzbarem Traktor) verarbeiten. Maximal sind drei Durchschläge möglich.

Informationen: Qume GmbH, Schiess-Straße 55
4000 Düsseldorf 11



adcomp X100/X100S

Vierfarb-Trommelplotter

Versehen mit einer Centronics-Schnittstelle, lassen sich die adcomp-Plotter X100 und X100S problemlos mit dem Schneider-Computer im Text- und Grafikmodus betreiben. Die Geräte ignorieren z.B. auch das Doppel-Linefeed (zweifacher Zeilenvorschub), das der Rechner sendet.

Die beiden Plotter verfügen im Grafikmodus über einen reichen Befehlssatz, z.B. für durchgezogene oder gestrichelte Linien, sechs Symbole für Schnittpunktmarkierung o.ä., für Kreise und Rechtecke, die wahlweise schraffiert werden können u.v.m.

Wahlweise können die Plotter mit Einzelblättern oder Endlosformularen "gefüttert" werden. Dank der außergewöhnlich guten Bedienungsanleitung ist die Einarbeitung in wenigen Stunden möglich.

Informationen: adcomp datensysteme, Olgastraße 15
8000 München 19

Sie besitzen einen Schneider-Computer und wollen nun wissen, wie Ihr Computer laufen lernt?

Dann schauen Sie einmal in dieses Buch. Sie finden wertvolle Anwenderprogramme, die Sie sonst teuer kaufen müssen. Aus dem Inhalt:

- Archiv Ordnung ohne Papierkrieg
- Bauen & Wohnen Finanzierung, Rendite,
Abschreibung
- Terminkalender Immer auf dem neuesten Stand
- KlarText Textverarbeitung mit Komfort
- Drucker-Zeichen Entwürfe nach eigenem
Geschmack
- Zeichensatz Umlaute und Sonderzeichen
- Balkendiagramm Bildverarbeitung in BASIC
- Angebots-Plotter Plotter-Einsatz für Profis
- Key-Maker Wunder auf Tastendruck
- Adressendatei Verwaltung mit System
- Entspannung Kleine Spielchen für
zwischen durch
- Peripheres Passender Anschluss für den CPC

Ein besonderer Vorteil dieses Buches: Alle Listings sind ausführlich kommentiert und können daher leicht verstanden werden.

Geometric Algebra

Clifford Algebra

Spinors

Geometric Algebra

AMSTRAD CPC



MÉMOIRE ÉCRITE
MEMORY ENGRAVED
MEMORIA ESCRITA



<https://acpc.me/>